


EXPERIMENTOS

CON TU

COMPUTADORA



CON MUCHOS
PROGRAMAS Y
EXPERIENCIAS

EXPERIMENTOS CON TU COMPUTADORA

Helen Davies



© 1985 Usborne Publishing

© 1987 Publicaciones y Ediciones Lagos, S.A. PLESA, Sestao, 1. Pinto (Madrid)

Printed in Spain. Impreso en España


MELSA, Pinto (Madrid)

Depósito Legal: M-14496-1988 I.S.B.N.: 84-7374-187-0

«No está permitida la reproducción total o parcial de este libro, ni su tratamiento informático, ni la transmisión de ninguna forma, o por cualquier medio, ya sea electrónico, mecánico, por fotocopia, por registro u otros métodos, sin el permiso previo y por escrito de los titulares del Copyright.»

Contenido


- 3 Sobre este libro
- 4 Computadoras en experimentos e investigación
 - 6 Uso de los programas de este libro
 - 7 Programa para jugar a cara o cruz
 - 8 Pelota que rebota
 - 10 Experimentos con pulsaciones
 - 14 El camino de Ethel
 - 20 Sobre modelos económicos
 - 21 Dirección de una línea aérea
 - 24 Experimentos con sensores
 - 32 Almacenamiento de información
 - 37 Observación de los resultados
 - 40 ¿Probable o improbable?
 - 42 ¿Conversiones para los programas?
 - 46 Rutinas de gráficos
 - 48 Indice




Sobre este libro

Este libro contiene gran cantidad de divertidos programas que te permitirán usar tu computadora para hacer experimentos y analizar información. Los programas son ejemplos sencillos de cómo los científicos, economistas y demás investigadores usan las computadoras. Están escritos para computadoras personales y funcionan en el Commodore 64, VIC 20, TRS-80 Colour Computer (32k), Apple II, BBC (B), Electron y Spectrum.*


En la página 8 hay un programa que representa una bola botando. Puedes hacer pruebas con ella y ver qué sucedería si rebotase la bola en otro planeta.



Algunos de los programas son por sí solos experimentos. Puedes cambiar algunos datos del programa y ver lo que sucede. Estos programas se denominan modelos. Te permiten comprobar teorías y hacer predicciones.



Otros programas están pensados para trabajar con los resultados que obtienes de experimentos y encuestas. Te ayudan a analizar los datos, así como a almacenarlos y visualizarlos de forma útil.



En las páginas 24-31 hay instrucciones paso a paso para fabricar sensores de temperatura y de luz para tu computadora. Los sensores son aparatos electrónicos que miden cosas como la temperatura o la luz y convierten las mediciones en señales eléctricas. Usando sensores la computadora puede realizar lecturas directamente de un experimento y contarte lo que está sucediendo.



En la página 6 puedes ver cómo escribir y usar los programas. En las páginas 46-47 hay una rutina de gráficos que puedes añadir a diferentes programas.

*Los sensores de las páginas 24-31 sólo funcionan con computadoras que poseen un port «analógico», es decir: Commodore 64, VIC 20, TRS-80 y BBC.

Computadoras en experimentos e investigación

Los programas que usan los científicos y demás investigadores suelen ser muy complejos y normalmente funcionan en computadoras muy potentes. En estas dos páginas hay dibujos que demuestran cómo se usan algunos programas de este tipo. Los programas de este libro son mucho más sencillos que los programas aquí descritos aunque más o menos realizan la misma función.

Reconstruir la gran explosión

Las computadoras se usan para reconstruir sucesos demasiado extensos o demasiado complejos para ser estudiados en la realidad. Por ejemplo, en EE.UU. los físicos están usando una de las computadoras más potentes que existen —una Cray 1— para reconstruir el nacimiento del Universo.

La reconstrucción se basa en las teorías de los físicos que establecen que el Universo surgió como consecuencia de una gigantesca explosión. El programa usa ecuaciones matemáticas para representar lo que sucedió durante el primer segundo de la explosión y la siguiente evolución hasta el universo que hoy conocemos.

Comparando las predicciones que pueden obtenerse de esta reconstrucción con las observaciones actuales, los científicos pueden ir verificando si sus teorías tienen posibilidades de ser ciertas.

Sensores en la investigación cerebral

Este es un dibujo de computadora de alguien pensando. Para obtenerlo se conectaron unos sensores denominados electrodos a la cabeza de una persona y se introdujeron todos los datos captados en una computadora.




Electrodos


Los diferentes colores muestran zonas de mayor o menor actividad eléctrica del cerebro. Dibujos como éste ayudan a los investigadores a descubrir para qué actividades se usan las diferentes zonas del cerebro. Se usan actualmente para averiguar por qué algunas personas tienen problemas de lectura, así como para diagnosticar tumores cerebrales y epilepsia.

En este libro averiguarás cómo conectar sensores de luz y de temperatura a tu computadora y cómo usarlos en tus experimentos.

Dibujo de computadora




Las computadoras crean modelos que usan ecuaciones matemáticas ya que realizan los cálculos a gran velocidad. En este libro encontrarás varios programas para crear modelos.



Diseñar un cohete


Los modelos hechos en computadoras sirven también para desarrollar y probar productos y máquinas como, por ejemplo, las naves espaciales. El dibujo inferior se obtuvo del modelo de un cohete despegando, producido por una computadora.



Las rayas verdes muestran el flujo de aire que se forma alrededor del cohete cuando éste viaja a casi la velocidad del sonido. Experimentando con el modelo, los ingenieros espaciales pudieron predecir el efecto que causaba el cambiar el ángulo de ataque del cohete (es decir, el ángulo con el que penetra en el espacio).

Resolución de delitos

Normalmente la investigación suele dar como fruto grandes cantidades de información. Los programas denominados bases de datos permiten a los investigadores almacenar la información en una computadora. La computadora puede buscar en segundos en una base de datos un dato o un detalle y lograr relaciones que de otra manera no se obtendrían. En las páginas 32-36 tienes un programa base de datos que te servirá para almacenar información de encuestas o experimentos.




Uso de los programas de este libro

Antes de usar los programas de este libro lee estos consejos para copiarlos y para ejecutarlos. Algunas líneas del programa tendrán que modificarse para las diferentes computadoras. Estas están señaladas con un asterisco y podrás saber

lo que tienes que cambiar mirando en las líneas de conversión que tienes en las páginas 42-45. Cuando hayas introducido un programa y compruebes que funciona, grábalo en cinta por si alguna vez quieres volver a usarlo.


1



Copia los programas tal y como están. Al final de cada línea comprueba que no existan errores y aprieta RETURN (ENTER en algunas computadoras).* Cuando encuentres una línea con un asterisco pasa a

las páginas 42-45 y busca la conversión correspondiente a tu computadora. Si no hay ninguna conversión copia la línea tal y como está en el programa.

2




Algunos programas necesitan una rutina de gráficos antes de poder usarlos. Existen distintas versiones de la rutina para cada computadora. Las puedes encontrar en las páginas 46-47. Asegúrate de añadir la que es correcta para tu computadora.

3

Casi todas las computadoras suelen dar un mensaje en el que se indica dónde está el error.


```
RUN
MISTAKE AT LINE 20
LIST
10 CLS
20 LFT H=0
```

LET está mal escrito



Para hacer que la computadora ejecute un programa escribe RUN. Si el programa no funciona a la primera, es posible que exista un error (denominado chinché). Para corregir el programa lista las líneas del programa en la pantalla y vuelve a escribir o editar las líneas con errores.

4



En ocasiones tendrás que añadir o modificar líneas. Para hacer esto tendrás que parar el programa que se está ejecutando. Aprieta la tecla «escape» (como se muestra arriba).

5

ESCAPE AT LINE 50

```
LIST
10 CLS
20 LET H=0
30 LET T=0
```

Luego, lista el programa e introduce las nuevas líneas con sus números de línea. Para volver a ejecutar el programa escribe otra vez RUN.

*Si tu computadora tiene una tecla ENTER apriétala cada vez que en este libro veas RETURN.

Programa para jugar a cara o cruz

Este es un programa sencillo que puedes probar en tu computadora. Simula el lanzamiento de una moneda. Puedes experimentar con él para ver qué resultados obtienes si tiras una moneda un número determinado de veces.

```
* 10 CLS
  20 LET H=0
  30 LET T=0
  40 PRINT:PRINT
  50 INPUT N
  60 FOR K=1 TO N
* 70 LET X=RND (1)
  80 IF X< 0.5 THEN LET H=H+1
  90 IF X>0.5 THEN LET T=T+1
 100 NEXT K
 110 PRINT "NUMERO DE CARAS = ";H
 120 PRINT "NUMERO DE CRUCES = ";T
```

Comprueba en las páginas 42-45 para ver si necesitas cambiar alguna de las líneas del programa señaladas con un asterisco.

"CUANTOS LANZAMIENTOS?"



Ejecución del programa

Cuando ejecutas el programa le dices a la computadora cuántas veces debe lanzar la moneda. Simula los lanzamientos y te dice cuántas caras y cuántas cruces se obtienen.

1 ¿CUANTOS LANZAMIENTOS?
?10

NUMERO DE CARAS = 8
NUMERO DE CRUCES = 2

2 ¿CUANTOS LANZAMIENTOS?
?100

NUMERO DE CARAS = 48
NUMERO DE CRUCES = 52

Simula el lanzamiento diez veces y fíjate en el resultado que se obtiene. Luego prueba cien o mil lanzamientos.

Observarás que cuantas más simulaciones de lanzamientos realices, más te acercarás al resultado de 50% caras y 50% cruces.

Más sobre modelos en computadoras

Aunque éste sea un programa muy sencillo demuestra cómo una simulación hecha con una computadora puede ahorrar tiempo y dinero a la hora de hacer experimentos reales.

Prueba a calcular cuánto tardas en lanzar una moneda cien veces y anotar los resultados. ¿Qué tal es tu tiempo comparado con el de la computadora?




Pelota que rebota

El programa de esta página simula una pelota botando. Puedes experimentar con él y ver lo que sucede si lanzas la pelota a diferentes alturas o con una fuerza diferente, o bien si usas una pelota que rebote más. Puedes incluso variar la fuerza de la gravedad.


Cuando ejecutes el programa tienes que dar a la computadora información sobre la pelota tal y como se indica a la derecha.

Altura inicial



Esta es la altura sobre el suelo de tu mano cuando sueltas la pelota. Puedes probar con 1 metro o 3 pies.

Dirección inicial




La dirección de la pelota se da con un ángulo, por ejemplo, en grados. Si lanzas la pelota hacia arriba el ángulo estará entre 0° y 90°.

Programa de la pelota que rebota

Copia estas líneas de programa y mira en las páginas 46-47 para añadir la rutina de gráficos que valga en tu computadora. El programa calcula la distancia en metros. Si quieres que use pies sustituye las líneas 20, 40, 50, 70 y 170 por las que se dan en la parte de abajo de la página.

```
* 10 CLS:PRINT
20 PRINT "ALTURA INICIAL (M)": INPUT H
30 PRINT "DIRECCION INICIAL (GRADOS)": INPUT A
40 PRINT "VELOCIDAD INICIAL (M/S)": INPUT V
50 PRINT "GRAVEDAD (M/S/S)": INPUT G
60 PRINT "CAPACIDAD DE REBOTE (0-1)": INPUT B
70 LET T$="": LET Y$="5 M": LET X$="20 M": LET M$="2.5"
80 GOSUB 2000
90 LET DT=0.01: LET TB=20: LET NB=0: LET K=3.14/180
100 LET HS=V*SIN(A*K): LET VS=V*COS(A*K)
110 LET NX=0: LET NY=H
120 LET PX=NX: LET PY=NY: LET US=VS
130 LET NX=PX+HS*DT
140 LET VS=US-G*DT
150 LET NY=PY+((US+VS)/2)*DT
160 IF NY<=0 THEN GOSUB 200
170 LET Y=PY*200: LET X=PX*50: LET N=0: GOSUB 2200
180 IF X<900 AND NB<TB THEN GOTO 120
190 STOP
200 LET VS=SGN(US)*SQR((ABS(US)^2)+2*G*PY)
210 LET HT=(US-VS)/G
220 LET NX=PX+HS*HT: LET NY=0
230 LET VS=-VS*B: LET NB=NB+1
240 RETURN
```

Recuerda que debes añadir las líneas de gráficos de tu computadora



```
20 PRINT "ALTURA INICIAL (PIES)": INPUT H
40 PRINT "VELOCIDAD INICIAL (PIES/S)": INPUT V
50 PRINT "GRAVEDAD (PIES/S/S)": INPUT G
70 LET T$="": LET Y$="15 PIES": LET X$="60 PIES": LET M$="7.5"
170 LET Y=PY*66: LET X=PX*17: LET N=0: GOSUB 2200
```


Velocidad inicial



5 m/s (15 pies/s) en un lanzamiento suave.

Esta es la velocidad de la pelota en el momento en que abandona tu mano. Se mide en metros por segundo (m/s) o pies por segundo (pies/s).


Gravedad



En la Tierra la gravedad es sobre 9,8 m/s/s (32 pies/s/s).

La gravedad es la fuerza que atrae los objetos a la tierra. Se mide en metros por segundo al cuadrado (m/s/s) o pies por segundo al cuadrado (pies/s/s).

Rebote



Esta es la medición de la capacidad de rebote que elijas. Puede ir entre 0 y 1. Una pelota dura de goma sería de 0,9. Prueba 0,5 ó 0,7.


Ejecución del programa

Altura inicial (M)
?1.5
Dirección inicial (Grad)
?45
Velocidad inicial (M/S)
?6
Gravedad (M/S/S)
?10
Capacidad de rebote
?0,7

Prueba a variar la altura inicial o la capacidad de rebote de la pelota.

Ejecuta el modelo con estos valores. ¿Parece una pelota rebotando? Ejecútalo nuevamente manteniendo todo igual menos la dirección. Prueba con 60°, 20°, 135°. ¿Cuál hace que la pelota rebote más alto?

Experimento en Mercurio



En el planeta Mercurio la gravedad es sólo de 3,7 m/s/s (12,1 pies/s/s). Nadie ha estado nunca allí pero puedes usar el modelo para averiguar lo que sucedería al rebotar una pelota en Mercurio. Haz que la gravedad sea 3,7 (o 12,1) y mantén el resto igual.

Cómo funciona el modelo

Como la reconstrucción de las páginas 4-5 este programa es un modelo matemático. Calcula la trayectoria de la pelota usando ecuaciones basadas en leyes universales de la física. Fueron descubiertas por un físico llamado Isaac Newton quien vivió entre 1642 y 1727. Las leyes se refieren al movimiento de los objetos y a la fuerza de la gravedad.



Guía del universo para el programa de la pelota botadora

VENUS *	8,8 m/s/s ó 28,9 pies/s/s
MARTE	3,7 m/s/s ó 12,1 pies/s/s
JUPITER	26,5 m/s/s ó 86,9 pies/s/s
SATURNO	11,8 m/s/s ó 38,7 pies/s/s
URANO	9,1 m/s/s ó 29,6 pies/s/s
NEPTUNO	12 m/s/s ó 39,4 pies/s/s
PLUTON	0,4 m/s/s ó 1,3 pies/s/s

Estos son los valores de la gravedad en otros planetas por lo que puedes llevar tu pelota a rebotar por el Universo.

Experimentos con pulsaciones


El programa de las dos páginas siguientes hace que la computadora almacene lecturas de tus pulsaciones y las visualice en un gráfico. Puedes usarlo para comparar tus pulsaciones antes y después de hacer ejercicio. En la página opuesta puedes encontrar más datos para ejecutar el programa. Debajo tienes algunos consejos para usarlo.



Existen varias formas diferentes de medir tus pulsaciones. Estas se te indican en la página 13. Pruébalas antes de empezar y usa la que te resulte más fácil.

Consejos para usar el programa

Usa el programa con una serie de lecturas de pulsaciones antes de empezar a realizar ejercicios. La computadora visualizará un gráfico de tus pulsaciones normales.



Ideas para ejercicios

JOGGING

HACER FLEXIONES

HACER ABDOMINALES

CICLISMO

SPRINTS DE CORTA DURACION

SUBIR ESCALERAS (SUBIENDO Y BAJANDO A UNA SILLA)

Mide la cantidad de ejercicio que haces. Es decir, haz por ejemplo jogging durante un período fijo de tiempo como pueden ser diez minutos, o corre una distancia determinada o cuenta el número de flexiones que haces.



La computadora trazará un segundo gráfico en un color diferente que te permitirá observar el aumento de tus pulsaciones y el tiempo que tardan en volver a la normalidad.

El programa te permite hacer una tercera serie de lecturas, por lo que puedes hacer un ejercicio diferente y comparar tu ritmo de recuperación para dicho ejercicio.

Ejecución del programa

1 APRIETA UNA TECLA CUANDO ESTES LISTO PARA EMPEZAR

Cuando ejecutes el programa este mensaje te dice que estés listo para tomar las pulsaciones. Una vez que aprietas una tecla tendrás unos 15 segundos para encontrar tu pulso.

2 GRUPO DE DATOS 1
TEST 1

COMIENZA LA CUENTA...
AHORA

Empieza a contar tus pulsaciones una vez que la computadora dé un pitido y visualice este mensaje. Después de 15 segundos la computadora da otro pitido que te indica que debes dejar de contar.

3 GRUPO DE DATOS 1
TEST 1

COMIENZA LA CUENTA...
AHORA
STOP

¿CUANTAS PULSACIONES? 22

Introduce el número de pulsaciones que contaste y aprieta RETURN. La computadora multiplica éste por cuatro para calcular tus pulsaciones por minuto.

4 GRUPO DE DATOS 1
TEST 2

COMIENZA LA CUENTA...
AHORA

Cuando hayas introducido la lectura tienes otros 15 segundos para volver a encontrar tu pulso. Comienza a contar cuando la computadora pite.



Después de tomar diez lecturas del pulso la computadora visualiza un gráfico mostrando el ritmo de pulsaciones por minuto contrastado con el tiempo.

6 GRUPO DE DATOS 2
TEST 1

COMIENZA LA CUENTA...
AHORA

Haz ejercicio y toma la segunda serie de lecturas (denominada GRUPO DE DATOS 2). Para hacer que la computadora vuelva a tomar mediciones aprieta nuevamente cualquier tecla.

Programa de pulsaciones

La parte principal del programa se da aquí abajo y en la página siguiente. Cópialo y añade la rutina de gráficos de las páginas 46-47.

```
10 GOSUB 500
20 GOSUB 460
30 FOR S=1 TO 3:GOSUB 210
40 LET X$="TIEMPO":LET Y$="150":LET M$="75"
50 LET T$="PULSACIONES / MIN":GOSUB 2000
```


El listado continúa en la página siguiente


```


60 FOR J=1 TO S
70 LET DS=J:GOSUB 120
80 NEXT J
90 GOSUB 390
100 NEXT S
110 GOSUB 390:STOP
120 LET DC=C(DS)
130 LET M$="GRUPO DE DATOS"+CHR$(48+DS):GOSUB 2400
140 FOR K=1 TO TN-1
150 LET DY=(N(DS,K+1)-N(DS,K))/10
160 FOR W=0 TO 9
170 LET X=K*K1+W*K2-K1:LET Y=(N(DS,K)+DY*W)*6.6
180 LET N=N(DS,K):GOSUB 2200
190 NEXT W:NEXT K
200 RETURN
210 GOSUB 420:PRINT:PRINT
220 PRINT "APRIETA UNA TECLA CUANDO"
230 PRINT "ESTES LISTO PARA EMPEZAR"
240 GOSUB 390
250 FOR K=1 TO TN
260 GOSUB 420:PRINT:PRINT
270 PRINT "GRUPO DE DATOS"; S
280 PRINT "TEST "; K:PRINT:GOSUB 370
290 PRINT "COMIENZA LA CUENTA...":GOSUB 350
300 GOSUB 440:PRINT "AHORA":GOSUB 370
310 GOSUB 440:PRINT "STOP":PRINT
320 PRINT "CUANTAS PULSACIONES ...:INPUT X
330 LET N(S,K)=X*4:NEXT K
340 RETURN
350 FOR T=1 TO PT:NEXT T
360 RETURN
370 FOR T=1 TO FT:NEXT T
380 RETURN
* 390 GET I$
400 IF I$="" THEN GOTO 390
410 RETURN
* 420 CLS
430 RETURN
440 PRINT CHR$(7)
450 RETURN
460 LET TN=10:LET K1=1000/(TN-1):LET K2=K1/10
470 DIM N(3,TN):DIM C(3)
* 480 LET C(1)=1:LET C(2)=2:LET C(3)=3
490 RETURN
* 500 LET FT=7285:LET PT=900
510 RETURN

```


Añade ahora la
rutina de gráficos
para tu computadora.



Formas de tomarte el pulso



El lugar más habitual para tomar el pulso es en la muñeca. Súbete la manga del brazo derecho y sujeta tu mano derecha con la palma hacia arriba. Pon el primero, el segundo y el tercer dedo de tu mano



izquierda sobre el extremo de la muñeca de tal forma que estén empujando la piel hacia dentro. Justo bajo la piel hay una arteria que te debe permitir notar el pulso bajo la punta de los dedos.




Otro lugar para tomar el pulso es en el cuello. Coloca los dedos de tu mano derecha sobre la zona saliente de la parte delantera. A continuación deslízalos unos 5 cm hacia la izquierda. Hay un punto hundido de la piel en el que puedes percibir las pulsaciones.




Si no puedes localizar tu pulso fíjate en si notas los latidos de tu corazón. Coloca tu mano sobre la tercera o cuarta costilla hacia arriba en la parte izquierda de tu cuerpo. En este punto probablemente notarás que el corazón late.

¿Te hallas en buena forma?

Las pulsaciones vuelven a su estado normal después de hacer ejercicio, más rápidamente cuanto más en forma estés. Los gráficos inferiores muestran el ritmo de recuperación para distintos niveles de forma física. Compara tu gráfico con éstos para ver en qué forma te hallas.



Si estás realizando ejercicios regulares puedes usar el programa para observar si logras mejoría en tu forma física. Repite los tests de pulsaciones con intervalos



superiores a varios meses y anota cada vez la forma de los gráficos. Comparando éstos podrás ver si tu ritmo de recuperación está mejorando.

El camino de Ethel

Ethel acaba de cambiarse de casa y quiere averiguar si es más cómodo ir al trabajo en tren o en autobús. El programa de la página opuesta puede ayudarla. Simula tantas veces como quieres los viajes en autobús y en tren, por lo que ella puede deducir cuál es el mejor, tras un período de seis meses o un año.

Los simuladores como éste se usan para

la investigación operacional, es decir, para encontrar la manera más adecuada de hacer cosas. Por ejemplo, ¿cómo debería organizarse un self-service para que la gente no tenga que hacer largas filas? o, ¿cuál es la mejor manera de distribuir el cargamento en los barcos, o el tráfico en las ciudades?

¿Autobús o tren?

Aquí hay alguna información sobre la trayectoria que tiene que seguir Ethel en tren y en autobús.

1 Si Ethel va en autobús tiene una caminata de tres minutos hasta la parada.

2 Los autobuses vienen cada siete minutos, pero los asiduos a este autobús dicen que uno de cada diez días se cancela un autobús.

3 El viaje en autobús debe durar unos 30 minutos, pero puede llegar hasta los 50 minutos si hay mucho tráfico.

4 Luego tiene una caminata de diez minutos hasta su oficina.

5 La caminata hasta la estación es de 13 minutos.

6 Los trenes salen a los 15 y a los 45 minutos de cada hora con un 78% que son puntuales y un 6% que llegan más de cinco minutos tarde. Ethel averigua además que uno de cada 50 días se cancela un tren.

7 El viaje en tren es de 23 minutos y hay una caminata desde la estación a la oficina de Ethel de ocho minutos.

Ethel ha probado ambos métodos varias veces, pero cada parte de la ruta cambia de un día a otro. En ocasiones el autobús se retrasa y en ocasiones se cancela el tren.

Para saber qué camino es mejor, tendrá que comparar un número muy alto de viajes en tren y en autobús. La única manera rápida de hacer esto es con la simulación por computadora.


Programa simulador de trayectos

Este programa simula ambas rutas, usando números aleatorios para admitir las variaciones que hay de un día a otro. Por ejemplo, en un día determinado la espera de Ethel en la parada del autobús puede ser entre 0 y 7 minutos. Por lo que cada vez que la computadora simula el viaje en autobús, toma un número aleatorio entre 0 y 7. Hace algo similar con las siguientes etapas del

camino y luego suma los tiempos para obtener la duración total de la ruta del autobús. Obviamente el tiempo de un solo viaje no es de mucha ayuda, pero si haces que la computadora simule el viaje un gran número de veces, obtendrás una idea de la duración media del viaje tras varios meses y podrás compararlo con la duración de la ruta del tren.

```
* 10 CLS
20 GOSUB 860:PRINT "INFORMACION SOBRE LA SIMULACION DEL CAMINO"
30 PRINT "(TIEMPO EN MINUTOS)"
40 PRINT "AUTOBUS":PRINT
50 PRINT "CAMINATA HASTA LA PARADA":INPUT W1B
60 PRINT "TIEMPO ENTRE LOS AUTOBUSES":INPUT FB
70 PRINT "DURACION MINIMA DEL VIAJE":INPUT SB
80 PRINT "DURACION MAXIMA DEL VIAJE":INPUT LB:LET DB=LB-SB+1
90 PRINT "PORCENTAJE DE AUTOBUSES CANCELADOS":INPUT PB
100 PRINT "CAMINATA DESDE LA PARADA":INPUT W2B
110 PRINT:PRINT "TREN":PRINT
120 PRINT "CAMINATA HASTA LA ESTACION":INPUT W3T
130 PRINT "HORARIO DE TRENES (HRS.MINS)":FOR I=1 TO 6
140 INPUT TT:GOSUB 840:LET T(I)=TT
150 NEXT I
160 PRINT "DURACION DEL VIAJE":INPUT TJ
170 PRINT "PORCENTAJE DE TRENES PUNTUALES":INPUT PT
180 PRINT "PORCENTAJE DE TRENES CON MAS DE 5 MINUTOS DE RETRASO":INPUT P5
190 PRINT "PORCENTAJE DE TRENES CANCELADOS":INPUT PC
200 PRINT "CAMINATA DESDE LA ESTACION":INPUT W4T
210 PRINT:PRINT
220 PRINT "HORA DE LLEGADA (HH.MM.) ":INPUT A$
230 LET TT=VAL(A$):GOSUB 840:LET TA=TT
240 PRINT "HORA DE PARTIDA (HH.MM.) ":INPUT L$
250 LET TT=VAL(L$):GOSUB 840:LET TL=TT
260 PRINT "CUANTOS VIAJES":INPUT NJ
270 FOR I=1 TO 2:FOR J=1 TO 3
280 LET L(I,J)=0
290 NEXT J:NEXT I
300 FOR K=1 TO NJ
310 LET TS=TL
* 320 LET OV=rnd*100
330 IF OV<2 THEN LET TS=TS+rnd*30
340 LET T=TS
350 GOSUB 610
360 LET T=TS
370 GOSUB 700
380 FOR I=1 TO 2
390 IF J(I)<=0 THEN LET L(I,1)=L(I,1)+1
```

La instrucción con la que se obtienen números aleatorios varía. En este programa está como «rnd». Al copiar el programa sustituye rnd por la correspondiente de tu computadora (págs. 42-45).




```

400 IF J(I)>0 AND J(I)<=5 THEN LET L(I,2)=L(I,2)+1
410 IF J(I)>5 THEN LET L(I,3)=L(I,3)+1
420 NEXT I
430 NEXT K
440 PRINT:PRINT "RESULTADOS:"
450 PRINT:PRINT "TRAS ";NJ;" RECORRIDOS"
460 PRINT "SALIENDO A LAS ";L$
470 PRINT "PARA LLEGAR A LAS ";A$
480 PRINT
490 FOR I=1 TO 2
500 IF I=1 THEN PRINT "EN AUTOBUS"
510 IF I=2 THEN PRINT "EN TREN"
520 PRINT
530 PRINT "A LA HORA "L(I,1)
540 PRINT "MENOS DE 5 MINS TARDE ";L(I,2)
550 PRINT "MAS DE 5 MINS TARDE ";L(I,3)
560 PRINT
570 NEXT I
580 PRINT "EJECUTAR EL PROGRAMA OTRA VEZ (S/N):INPUT A$
590 IF A$="S" THEN GOTO 210
600 STOP
610 LET T=T+W1B
* 620 IF rnd*100<PB THEN LET T=T+FB
* 630 LET WT=INT(rnd*FB)
640 LET T=T+WT
* 650 LET BJ=SB+INT(DB*(rnd+rnd)*0.5)
660 LET T=T+BJ
670 LET T=T+W2B
680 LET J(1)=T-TA
690 RETURN
700 LET T=T+W3T
710 LET I=0
720 LET I=I+1:IF I=7 THEN PRINT "NO MAS TRENES":STOP
730 IF T(I)<T THEN GOTO 720
740 LET TT=T(I)
* 750 IF rnd*100<PC THEN LET TT=T(I+1)
* 760 LET DL=0:LET R=rnd*100
* 770 IF R>PT THEN LET DL=INT(rnd*5)
* 780 IF R>(100-P5) THEN LET DL=INT(rnd*20+5)
790 LET T=TT+DL
800 LET T=T+TJ
810 LET T=T+W4T
820 LET J(2)=T-TA
830 RETURN
840 LET TT=INT(TT)*60+INT(100*(TT-INT(TT))+0.5)
850 RETURN
860 DIM L(2,3):DIM J(2):DIM T(6)
870 RETURN

```


Uso del programa

Cuando ejecutes el programa tendrás que introducir la información sobre ambas rutas tal y como se muestra a la derecha. La información sobre la ruta de Ethel viene en el dibujo de la página 14.

En la página siguiente tienes algunos consejos para que veas el programa y simules tus propias rutas.



1 INFORMACION SOBRE LA SIMULACION DEL CAMINO (TIEMPO EN MINUTOS)

AUTOBUS
CAMINATA HASTA LA PARADA
?3
TIEMPO ENTRE LOS AUTOBUSES
?7
DURACION MINIMA DEL VIAJE
?30
DURACION MAXIMA DEL VIAJE
?50
PORCENTAJE DE AUTOBUSES CANCELADOS
?10
CAMINATA DESDE LA PARADA
?10

2

CANCELACION DE AUTOBUSES=1 de 10 ó 1/10
 $1/10 \times 100 = 10\%$

↖ Multiplica por 100 para convertir en porcentaje.

CANCELACIONES DE TRENES=1 de 50 ó 1/50
 $1/50 \times 100 = 2\%$

Las cancelaciones de trenes y autobuses han de darse en porcentajes. Aquí puedes ver cómo convertir cifras en porcentajes.

3 HORARIO DE TRENES (HRS.MINS)

?7.45
?8.15
?8.45
?9.15
?0
?0

Puedes introducir hasta seis horas para los trenes. Elige los trenes próximos a la hora en que piensas que Ethel debe salir. Si no quieres introducir seis, tienes que anular las últimas introduciendo un 0 y apretando RETURN.

4

HORA DE LLEGADA (HRS.MINS)
?9.00
HORA DE SALIDA (HRS.MINS)
?8.05

También tendrás que introducir la hora aproximada a la que quieres llegar y salir. Calcula la hora a la que debe salir Ethel para llegar al trabajo a las 9 en punto. Prueba con 8.00 ó 8.05 a.m.

5 CUANTOS VIAJES ?100

10?....100?....1000?

Diez pruebas no son más que dos semanas de trabajo —no son suficientes para calcular con precisión—. Mil, por otra parte, implicaría un cálculo demasiado largo.

La computadora debe simular suficientes viajes para que pueda darte una visión precisa de la duración de los viajes. Cien simulacros son cerca de seis meses de viajes y pueden darte una solución fiable.



Estudiar los resultados

Estos son algunos resultados obtenidos con el trayecto de Ethel. Tus resultados pueden no coincidir ya que las simulaciones varían cada vez que se ejecuta el modelo.

1 RESULTADOS

TRAS 100 RECORRIDOS
SALIENDO A LAS 8.05
PARA LLEGAR A LAS 9.00
EN AUTOBUS
A LA HORA 0
MENOS DE 5 MINS TARDE 31
MAS DE 5 MINS TARDE 30

EN TREN
A LA HORA 0
MENOS DE 5 MINS TARDE 0
MAS DE 5 MINS TARDE 100



Comparando los resultados del tren y del autobús puedes decidir qué medio de transporte es más adecuado para la hora de salida que diste.

2 RESULTADOS


TRAS 100 RECORRIDOS
SALIENDO A LAS 8.00
PARA LLEGAR A LAS 9.00
EN AUTOBUS
A LA HORA 63
MENOS DE 5 MINS TARDE 25
MAS DE 5 MINS TARDE 12

EN TREN
A LA HORA 93
MENOS DE 5 MINS TARDE 2
MAS DE 5 MINS TARDE 5

¿Por qué una diferencia de unos pocos minutos, al salir, crea una diferencia tan grande a la hora de llegar en tren?

¿Pero es la conclusión la misma si varías la hora de salida de Ethel? Ejecuta nuevamente el modelo alterando la hora de salida.

1 Experimentando



Supón que al jefe de Ethel no le importa que de vez en cuando llegue algo tarde pero que no consiente que llegue muy tarde ni que llegue siempre tarde. ¿A qué hora debe salir Ethel? ¿Debe ir en tren o en autobús?

2

SPROCKET & GIMBOL




Ethel está a cargo de abrir la oficina, por lo que nunca debe llegar tarde. En este caso ¿qué medio de transporte debe usar? ¿A qué hora debe salir?

Simula tus trayectos


Para encontrar información para el modelo fíjate en los horarios de trenes y autobuses, y llama al servicio local de transportes para preguntar sobre la duración de los trayectos y las cancelaciones. Pregunta también a personas que veas en las paradas y en la estación.

Si lo deseas simula una sola ruta. Para saltarte la otra ruta aprieta RETURN cada vez que la computadora te pida información sobre ella.



Más sobre simulaciones

En aquellas simulaciones que usen números aleatorios es importante que los números obtenidos den una gama realista de valores. Por ejemplo, el viaje en autobús de Ethel puede durar entre 30 y 50 minutos, pero normalmente es de 40 minutos. En el programa la instrucción puede obtener números aleatorios de los que la mayoría sean números próximos a 40.



Por otra parte, la espera de Ethel en la parada del autobús tiene tantas posibilidades de ser de dos minutos como de ser de cuatro o siete minutos, por lo que la instrucción para obtener este número da una gama de valores de este estilo.

1 Algo que probar

```
30
31 ****
32 ****
33 **
34 ****
35 ****
36 ****
37 *****
38 *****
39 *****
40 *****
41 *****
42 *****
43 *****
44 *****
45 *****
46 *****
47 ***
48 *
49 *
50
```

El programa inferior genera 100 números aleatorios para el viaje de Ethel en autobús. Los visualiza en la pantalla con tiras de estrellitas por lo que puedes observar la variedad de valores. Prueba a ejecutar varias veces el programa y observa el dibujo que se forma.

Programa del viaje en autobús


```
10 LET SB=30:LET DB=21
20 LET L=21:LET B=29
30 DIM F(L)
40 FOR N=1 TO 100
* 50 LET BJ=SB+INT(DB*(rnd+rnd)*0.5)
60 LET P=BJ+1-SB
70 LET F(P)=F(P)+1
80 NEXT N
* 90 CLS
100 PRINT:PRINT
110 FOR N=1 TO L
* 120 PRINT TAB(0);B+N;TAB(5);" ";
130 IF F(N)=0 THEN GOTO 150
* 140 FOR J=1 TO F(N):PRINT "*";NEXT J
150 PRINT:NEXT N
```

2

```
0-1 *****
1-2 *****
2-3 *****
3-4 *****
4-5 *****
5-6 *****
6-7 *****
```

Si sustituyes las líneas 10, 20, 50, 60 y 120 por las del cuadro, el programa generará la espera de Ethel en la parada del autobús. Compara esta gama de valores con la que obtienes con los tiempos del viaje en autobús.

El TRS-80 sólo puede visualizar la mitad de los tiempos para el viaje en autobús. Aprieta cualquier tecla para ver el resto.



Cambios para la espera

```
10 LET FB=7
20 LET L=7:LET B=-1
* 50 LET WT=INT(rnd*FB)
60 LET P=WT+1
* 120 PRINT TAB(0);B+N;
"-";B+N+1;" ";
```


Sobre modelos económicos

La economía es el estudio de cómo se produce y distribuye la riqueza en un país. La economía de un país es realmente compleja. La única manera por la que un economista puede tener en cuenta todos los factores que influyen es creando un modelo económico en una computadora.

1 Los gobiernos usan modelos económicos para predecir la inflación y el paro así como el efecto de sus políticas de subida de impuestos. Los modelos también sirven para demostrar lo que hubiese sucedido en el pasado si sus políticas hubiesen sido diferentes.

2 Cada país depende de una manera o de otra de sus exportaciones e importaciones. Un modelo económico debe tener en cuenta el hecho de que la economía de un país puede afectar a la de otro.

Algunos países producen y exportan bienes.

Los recursos naturales como el petróleo, el estaño o el oro son importantes para algunas economías.

3 Los programas que simulan la economía de un país se denominan modelos macro-económicos (macro significa a gran escala). Los modelos micro-económicos se usan para estudiar una compañía o una industria determinada. Ayudan a las empresas a planificar lo que deben hacer.

Las líneas aéreas y navales son servicios que los países pueden realizar por todo el mundo.

4 En ocasiones el modelo de una empresa está relacionado con un modelo macro-económico con el fin de averiguar las influencias sobre la empresa de los factores económicos generales.

Más sobre modelos económicos

Los modelos económicos se basan en teorías. Nadie entiende realmente cómo funciona la economía y muchos sucesos económicos nunca han sido realmente explicados.

Las diferentes tendencias políticas usan modelos distintos y, a menudo, discrepan sobre los efectos que tendrá una determinada política.

Dirección de una línea aérea

El programa inferior es un modelo económico muy sencillo para dirigir una compañía de líneas aéreas. Puedes usarlo para experimentar con diferentes estrategias empresariales y ver cuál funciona mejor.

```
10 LET G$=" G.".LET E$="DEMASIADOS"
20 DIM A(5)
30 LET AC=50000
40 LET RC=10000
50 LET M=500000
60 LET A=0
70 LET YR=1
80 LET TC=M.LET VA=0
90 GOSUB 880
100 LET P$="PREDICCIONES MACRO-ECONOMICAS PARA LOS PROXIMOS 5 AÑOS**"
110 GOSUB 800
120 PRINT "TASA DE INFLACION (%)"
130 GOSUB 860.LET FR=RT
140 PRINT "TIPOS DE INTERES (%)"
150 GOSUB 860.LET TR=RT
160 LET P$="TIPO DE CAMBIO**1. AUMENTADO*2. CONSTANTE*3. DECRECIENTE*"
170 GOSUB 800
180 INPUT ER:IF ER<1 OR ER>3 THEN GOTO 180
190 LET ER=0.8+ER/10
200 GOSUB 880
210 LET P$="POLITICA PARA EL PROXIMO AÑO**":GOSUB 800
220 PRINT "AÑO ";YR
230 LET X=0.LET P$="*COMENZARAS EL AÑO CON:**":GOSUB 800
240 GOSUB 740
250 LET P$="*CUANTOS AVIONES QUIERES COMPRAR*":GOSUB 800
260 INPUT NA:IF NA<0 OR NA*AC>M THEN PRINT E$.GOTO 260
270 LET A(YR)=NA*AC.LET CS=NA*AC
280 LET P$="*CUANTA TRIPULACION QUIERES CONTRATAR*":GOSUB 800
290 INPUT NC:IF NC<0 OR NC*RC+CS>M THEN PRINT E$.GOTO 290
300 LET CS=CS+RC*NC
310 LET P$="*CUANTO QUIERES GASTAR EN PUBLICIDAD*"
320 GOSUB 800
330 INPUT MV:IF MV<0 OR MV+CS>M THEN PRINT E$.GOTO 330
340 LET CS=CS+MV
350 GOSUB 880
360 LET P$="RESULTADOS DEL AÑO "+STR$(YR)+"**":GOSUB 800
370 LET X=TC.GOSUB 900.LET SC=TC
380 LET P$="COMIENZO DEL AÑO"
390 GOSUB 730.LET A=A+NA
400 LET BC=1000+2000*A
410 LET CS=CS+BC
420 LET X=CS.GOSUB 900
430 LET P$="*GASTO DE DINERO = "+STR$(X)+G$+"*":GOSUB 800
440 LET M=M-CS:IF M<0 THEN PRINT "ESTAS EN BANCARROTA!":STOP
450 LET F=NC/2:IF A<F THEN LET F=A
460 LET LF=1.4+(MV/(RC*F*2+1))*3
470 IF LF>2 THEN LET LF=2
480 LET CR=RC*F*LF*ER*2
490 LET X=CR.GOSUB 900
500 LET P$="DINERO RECIBIDO= "+STR$(X)+G$+"*":GOSUB 800
510 LET IN=INT(M*TR)
520 LET X=IN.GOSUB 900
530 LET P$="INTERESES POR EL DINERO INVERTIDO= "+STR$(X)+G$+"*":GOSUB 800
540 LET M=M+IN+CR
550 LET VA=0
560 FOR I=1 TO YR
570 LET A(I)=A(I)*0.8
580 LET VA=VA+A(I)
```


El listado continúa en la página siguiente.


```

590 NEXT I
600 LET AC=AC*(1+FR):LET RC=RC+(1+FR)
610 LET TC=VA+M
620 LET X=TC:GOSUB 900:LET P$="FINAL DEL AÑO":GOSUB 730
630 LET DC=TC-SC:LET P$="BENEFICIO":IF DC<0 THEN LET P$="PERDIDA"
640 LET X=ABS(DC):GOSUB 900:LET P$="*"+P$+"="+STR$(X)+G$+"*":GOSUB 800
650 LET Q$="PRINCIPIO":LET R$="FINAL"
660 IF DC<0 THEN LET Q$="FINAL:LET R$="PRINCIPIO"
670 LET P$="(CAPITAL AL "+R$+" DEL AÑO*- CAPITAL AL "+Q$+" DEL AÑO)"
680 GOSUB 800
690 PRINT:PRINT:PRINT "APRIETA RETURN"
700 INPUT X$
710 LET YR=YR+1:IF YR<6 THEN GOTO 200
720 PRINT:PRINT:STOP
730 LET P$="CAPITAL AL "+P$+"*":GOSUB 800
740 LET X=M:GOSUB 900:PRINT ""X,G$: " DINERO"
750 LET X=VA:GOSUB 900
760 LET P$="AVION VALE "+STR$(X)+G$+"*":GOSUB 800
770 LET X=VA+M:GOSUB 900
780 LET P$="TOTAL = "+STR$(X)+G$+"*"
790 GOSUB 800:RETURN
800 FOR L=1 TO LEN(P$)
810 LET M$=MID$(P$,L,1)
820 LET PS=POS(0)-28
830 IF (M$=" "AND PS>0) OR M$="*" THEN PRINT:GOTO 850
840 PRINT M$:
850 NEXT L:RETURN
860 INPUT RT:IF RT<0 OR RT>50 GOTO 860
870 LET RT=RT/100:RETURN
880 CLS
890 PRINT:PRINT:RETURN
900 LET X=INT(X/1000+0.5)*1000
910 RETURN

```


El programa utiliza una moneda imaginaria denominada Grotos (G para abreviar).



Sobre el programa

Controlas una pequeña compañía de líneas aéreas y estás planeando tu política para los cinco próximos años. Para empezar le das a la computadora información sobre las condiciones macro-económicas generales que pueden afectar a tu negocio. Estas son

la predicción de la inflación, los tipos de interés y el tipo de cambio. Debajo puedes encontrar más cosas sobre éstos. A continuación debes decidir las estrategias para cada año.



Tasa de inflación


La tasa de inflación (el ritmo al que suben los precios) afecta al coste a la hora de comprar un nuevo avión, pagar a la tripulación o pagar los gastos generales.

Tipos de interés

El tipo de interés determina cuánto puedes ganar por el dinero que guardas en el banco.

Tipos de cambio

El tipo de cambio indica cuántas divisas obtienes como intercambio por tu dinero. Ello afecta al número de billetes que vendes en el exterior. Un tipo de cambio en aumento significa que vendes menos billetes ya que resultan más caros.



Ejecución del programa

1 Predicciones macro-económicas

Introduce la inflación, el tipo de interés y el tipo de cambio previstos para cada año. Puedes inventar la información o bien acudir a cifras reales que puedes tomar de los periódicos.

PREDICCIONES MACRO-ECONOMICAS PARA LOS PROXIMOS 5 AÑOS

TASA DE INFLACION ANUAL (%)

?8

TIPO DE INTERES ANUAL (%)

?10

TIPO DE CAMBIO

1 AUMENTADO

2 CONSTANTE

3 DECRECIENTE

?1

2 Decidir el tipo de política

POLITICA PARA EL PROXIMO AÑO AÑO 1

COMENZARAS EL AÑO CON

500000 G. DINERO

AVION VALE 0 G.

Total = 500.000 G.

CUANTOS AVIONES QUIERES COMPRAR

?3

CUANTA TRIPULACION QUIERES CONTRATAR

?6

CUANTO QUIERES GASTAR EN PUBLICIDAD

?50000

1. Al comienzo de los cinco años un avión vale 50.000 G. y el precio asciende cada año según la tasa de inflación.

2. Necesitas dos tripulantes para cada avión. Los sueldos y costes de mantenimiento de un avión durante un año ascienden a 10.000 G. al principio de los cinco años. Estos costes aumentan cada año según la tasa de inflación.

3. Existe un gasto anual fijo de 1.000 G., más 2.000 G. por cada avión, que deben pagarse aunque no se usen los aviones.

4. Obtienes intereses al tanto por ciento establecido con todo el dinero que no gastes.

Estas son las decisiones que debes tomar cada año. A la derecha tienes información sobre los costes y demás factores que pueden afectar a tus decisiones.

3 Resultados

Cada año la computadora calcula tu capital (es decir, tu dinero en efectivo más el valor de los aviones), al principio y al final del año. También calcula el beneficio o pérdida que tengas.

Ejecuta varias veces el programa manteniendo la tasa de inflación, el interés y el cambio fijos, pero aplicando distintas políticas. Luego mantén iguales las políticas y varía la inflación o el tipo de cambio.

RESULTADOS DEL AÑO 1

CAPITAL AL PRINCIPIO DEL AÑO

500.000 G. DINERO

AVION VALE 0 G.

TOTAL = 500.000 G.

Este es el dinero que gastas en aviones y en la tripulación.

GASTO DE DINERO = 267.000 G.

DINERO RECIBIDO = 108.000 G.

INTERESES POR EL DINERO

INVERTIDO = 23.000 G.

Este es el dinero que obtuviste vendiendo billetes.

CAPITAL AL FINAL DEL AÑO

364.000 G. DINERO

AVION VALE 120.000 G.

TOTAL = 484.000 G.

Este es el interés por el dinero que no gastaste.

Los aviones pierden una quinta parte de su valor cada año.

PERDIDA = 16.000 G.

(CAPITAL AL PRINCIPIO DEL AÑO
- CAPITAL AL FINAL DEL AÑO)

Experimentos con sensores

En las páginas siguientes descubrirás cómo fabricar sensores que permitirán a tu computadora visualizar gráficos de cambios luminosos o de temperatura. Puedes usar los sensores en multitud de experimentos como pueden ser los cambios de temperatura de tu cuerpo al hacer ejercicio, al dormir o al estar enfermo. Debajo tienes instrucciones para fabricar un sensor de temperatura usando un componente electrónico denominado resistencia térmica. En las páginas 30-31 puedes encontrar instrucciones para fabricar el sensor luminoso.



Los sensores sólo funcionan en computadoras con port analógico, es decir, el Commodore 64, el VIC 20, el TRS-80 y el BBC

Conexion

Port analógico

Resistencia térmica

Una resistencia térmica permite, según la temperatura que tenga, que a través suyo fluyan diferentes cantidades de corriente eléctrica que la computadora puede

reconocer. Tú debes programar la computadora para que traduzca las señales eléctricas a lecturas de temperatura.

Cosas que necesitas

Estos son los materiales que necesitas para fabricar el sensor de temperatura. Puedes comprar las resistencias, la resistencia térmica y gran parte de los materiales en

tiendas especializadas o bien, pedirlos a un distribuidor electrónico. Busca la dirección de un distribuidor en alguna revista de electrónica.

Enchufe para conectar el port analógico de tu computadora.






Commodore 64 y VIC 20
Hembra de 9 patillas tipo D (entrada).



BBC
Macho para 15 patillas tipo D (clavija).



TRS-80
Clavija DIN de 6 patillas.




Resistencia térmica tipo VA1067S (si no consigues ésta, valdrá cualquier resistencia térmica de 100K).

Para el Commodore 64, VIC 20 y BBC — una resistencia de 1M ohmio (Ω), 0,25 ó 0,5 watos. Esta posee franja marrón, negro y verde.

Para el TRS-80 — una resistencia de 100 K ohmios, 0,25 ó 0,5 watos. Esta posee franjas en colores marrón, negro, amarillo.


Fabricación del sensor

En el dibujo de la derecha puedes ver la apariencia que tendrá tu sensor cuando esté terminado. Tendrás que soldar cables a las patillas por detrás de la clavija para a continuación unir los componentes a los cables. En la página 29 encontrarás algunos consejos para soldar. Los diagramas inferiores muestran a qué patillas deben soldarse los componentes. Asegúrate de que sigues el diagrama correspondiente a tu computadora.




Diagramas de los cables


VIC 20 y Commodore 64



BBC




TRS-80



Consejos


1. Usa cables largos (sobre 60 cm.) para conectar la resistencia térmica a la clavija. Esto es para que puedas separarla de la computadora cuando hagas experimentos. Puedes usar cables mucho más cortos para la resistencia.
2. En primer lugar suelda los cables a las patillas por detrás de la clavija y finalmente suelda la resistencia y la resistencia térmica a los cables. Comprueba cuidadosamente que cada cable está conectado a la patilla correcta.
3. Cuando hayas terminado el sensor, cubre todas las juntas y cables pelados con cinta adhesiva, ya que si no los cables pueden entrar en contacto provocando un cortocircuito que puede llegar a dañar tu computadora.



Comprobar la resistencia térmica

En esta página tienes un test con el que comprobar si la resistencia térmica funciona. Para ello debes empezar por apagar la computadora y conectar la resistencia térmica al port analógico. A la derecha puedes averiguar qué port debes usar. Luego conecta nuevamente la computadora y copia el siguiente programa.


En el VIC 20 el port analógico se denomina «control port», y en Commodore 64 «control port 1». Usa el port «right joystick» en el TRS-80 y en el BBC el denominado «analogue».



Programa para la resistencia térmica


```
10 GOSUB 500
20 GOSUB 1000
30 PRINT "R=";R
40 GOTO 20
500 LET K1=4000
510 LET K2=1
520 RETURN
* 1000 LET TR=1.84*PEEK(54297)
* 1010 LET R=1/(1/TR-1/1000)
* 1020 LET T=1/((LOG(R/K2))/K1+1/273)-273
1030 RETURN
```

Estas líneas son válidas sólo para el Commodore 64. Mira en las páginas 42-45 para modificarlas y que valgan para el VIC 20, TRS-80 y BBC.



1 R=343.011723
R=341.912446
R=342.017782
R=342.560132
R=342.980561
R=343.765498
R=344.912334
R=345.201673
R=345.780231
R=345.472109
R=344.826075
R=344.396664
R=343.756103

Tus números pueden ser diferentes a estos.



2

R=203.557312
R=203.825601
R=203.763328
R=204.144267
R=207.352971
R=209.788356
R=212.659770
R=215.998342
R=218.001539



Cuando ejecutes el programa saldrá una serie de números en la pantalla. Estos son producidos por las señales de la resistencia térmica que denominamos R.

Para comprobar que las señales vienen realmente de la resistencia térmica sujétala junto a algo caliente. Los valores de R

deben descender para posteriormente volver a subir cuando la alejes del calor. Si esto no sucede, el sensor no está funcionando. Comprueba que todas las conexiones sean correctas y estén firmes y que no haya cables pelados en contacto.

Calibrar la resistencia térmica

Antes de poder usar la resistencia térmica en experimentos, tendrás que convertir los valores R en temperaturas. Esto se

denomina calibrar la resistencia térmica. Debajo puedes encontrar cómo calibrarla entre las lecturas 0°C y 80°C.

1

R=552.025385
R=552.987102
R=550.721631
R=551.026710
R=550.724553
R=551.026019

El valor medio de R es 551.027

Pon la resistencia térmica en una pequeña bolsa de plástico o protégela con papel de celofán para que no se moje.



Para saber cuál es la lectura de la resistencia térmica a 0°C colócala en un plato con hielos. Asegúrate de que esté totalmente cubierta de hielo. Los valores de R en la pantalla ascienden. A los diez minutos llegarán a un nivel estable (aunque

siempre fluctuarán ligeramente). Aprieta ESCAPE para pasar el programa. De los números visualizados en la pantalla toma el valor más habitual de R y redondéalo hasta que tengas sólo 3 decimales.

2

510 LET K2= ←

30 PRINT "TEMPERATURA = ";INT (T)

Lista el programa y vuelve a escribir la línea 510 haciendo que K2 sea igual a tu valor de R. K2 es la mínima dentro de la escala de temperaturas.

Escribe aquí tu valor de R

A continuación sustituye la línea 30 por la versión dada arriba. Esta hace que la computadora visualice las lecturas de la temperatura en lugar de R.

3

TEMPERATURA = 55

Ten cuidado, la temperatura a 80°C puede quemarte.



4

TEMPERATURA = 78

Si la lectura de la temperatura de la computadora es muy baja, haz K1 menor. Si es muy alta haz K1 mayor.



Para saber la lectura de la resistencia térmica correspondiente al máximo de la escala, colócala en agua calentada a 80 °C. Puedes usar un termómetro de cocina para calcular la temperatura del agua. Debes hacer que las lecturas de temperatura de la computadora coincidan con las del termómetro sustituyendo el valor de K1 en la línea 500 del programa.

Para cambiar K1, para el programa y listalo en la pantalla. Luego experimenta haciendo K1 mayor o menor en docientos o trescientas unidades y ejecutando el programa para comprobar las nuevas lecturas de temperatura de la computadora. Varía K1 hasta que las lecturas de la computadora coincidan con las del termómetro.

Hacer experimentos


Aquí tienes algunas ideas sobre experimentos que puedes llevar a cabo con la resistencia térmica.

Puedes hacer que la computadora, en lugar de simples lecturas, visualice un

gráfico con sólo añadir las líneas de programa de la parte de abajo de esta página y la rutina de gráficos de las páginas 46-47.

Ideas para llevar a cabo

1. Toma la temperatura de tu cuerpo. Para ello sujeta fuertemente la resistencia térmica con la palma de la mano o bien colócala en tu axila.
2. Sitúa la resistencia térmica en una taza de agua caliente situada en una corriente de aire frío y observa la forma de la curva de enfriamiento que obtendrás en la computadora. Recuerda que debes colocar la resistencia en una bolsa de plástico para que no se moje.
3. Usa la resistencia térmica para medir los cambios de temperatura que tienen lugar al realizar algún experimento químico. Por ejemplo, mezcla una cucharada de vinagre con cuatro o cinco trozos de bicarbonato y mide los cambios de temperatura al reaccionar el ácido con la base.
4. Traza un gráfico con las variaciones de temperatura de una habitación durante un día o una noche completas. Para hacer esto, la computadora debe trazar el gráfico muy lentamente. Debajo puedes averiguar cómo cambiar el programa para que haga esto.



Líneas de gráficos

```
* 30 LET FT=10:LET DX=10
40 LET T$="TEMPERATURA":LET Y$="100":LET X$="TIEMPO"
50 LET M$="50":GOSUB 2000
60 LET X=0
70 LET SM=0:FOR J=1 TO 5
80 GOSUB 1000:LET SM=SM+T
90 NEXT J:LET T=SM/5
100 LET Y=T*10:LET N=INT(T):GOSUB 2200
110 FOR L=1 TO FT:NEXT L
120 LET X=X + DX:IF X<1000 THEN GOTO 70
130 LET M$="APRIETA UNA TECLA":GOSUB 2400
* 140 GET IS
150 IF IS=" " THEN GOTO 140
160 GOTO 40
```

Comprueba en las páginas 42-45 si debes hacer algún cambio para tu computadora.

Recuerda que debes añadirle la rutina de gráficos de las páginas 46-47.

El gráfico muestra cómo la temperatura cambia con el tiempo.* La computadora completa el gráfico en un minuto. Para hacer que realice uno nuevo, aprieta RETURN.

28 Para algunos experimentos necesitas que la computadora dibuje el gráfico más


lentamente para que, por ejemplo, muestre las variaciones que tienen lugar durante 10 minutos o durante 12 horas. Para que haga esto lista el programa e incrementa el valor de FT en la línea 30.

*En el TRS-8C el gráfico sólo muestra cambios de temperatura de 3°C o más.

Consejos para soldar



Enchufa el soldador y deja que se caliente. Asegúrate de que lo colocas en un sitio adecuado ya que la punta se pone muy caliente. Sujeta la clavija con cinta a una superficie fija para que puedas soldar.




Corta dos trozos de cable de 60 cm para la resistencia térmica y otros dos trozos más cortos para la resistencia. En los extremos corta 1 cm. de aislante, retuerce los filamentos de cable pelado y «estáñalos» (ver abajo).




Para soldar un cable a una patilla sujétala dentro o junto al extremo trasero de la patilla.




Toca el estaño con el soldador hasta que se forme una gota de estaño fundido en la punta.



A continuación coloca la punta del soldador en la junta y espera hasta que la gota de estaño caiga sobre ésta y se solidifique uniendo el cable a la patilla.




Para unir un componente (ej. la resistencia térmica) a los cables, retuerce juntos la patilla del componente y el cable tal y como se muestra arriba. A continuación, suéldalos.



Cuando hayas hecho todas las conexiones cubre todos los cables pelados con cinta aislante para evitar cortacircuitos. La forma más fácil es doblar la cinta a lo largo sobre los cables.

Cómo estañar el cable

Estañar significa cubrir un cable con una fina capa de estaño que una los pelillos. Esto hace que luego sea más fácil de soldar. Para estañar un cable retuerce los pelillos juntos. Luego pasa por encima el soldador dejando una fina película de estaño.




Fabricación de un sensor luminoso

Puedes convertir el sensor de temperatura en un sensor luminoso si sustituyes la resistencia térmica por una LDR. Una LDR es una resistencia dependiente de la luz. Es un componente electrónico que permite que por su interior fluyan cantidades variables de corriente eléctrica dependiendo de la luz que recibe.


En la página opuesta tienes un programa para el sensor luminoso. También debes añadir la rutina de gráficos de las páginas 46-47.

Cómo fabricar el sensor


Debajo puedes ver cómo debes quitar la resistencia térmica y soldar una LDR en su lugar. Si quieres fabricar un sensor luminoso nuevo compra otra clavija y otra




resistencia y fábricalo siguiendo las instrucciones que se dan para el sensor de temperatura en las páginas 24-25.




Para soltar la resistencia térmica, coloca el soldador caliente junto a la junta entre la resistencia y el cable. Al fundirse el estaño tira de la patilla de la resistencia.



Suelda la LDR en lugar de la resistencia térmica y cubre de cinta aislante las nuevas juntas.



Una LDR capta la luz que tiene alrededor. El sensor luminoso es más útil si sólo capta la luz en la dirección que tú apuntes. Para lograr esto pega un tubo de cartulina oscura alrededor de la LDR.




La longitud del tubo determina el tamaño de la zona que el sensor luminoso «lee». Cuanto más largo sea el tubo menor será esta área por lo que el sensor será más sensible.

Programa para el sensor luminoso

Para usar el sensor luminoso introduce las líneas inferiores y añade la rutina de gráficos de tu computadora (ver páginas 46-47).

```
* 10 LET FT=10:LET DX=10
20 LET T$="INTENSIDAD DE LUZ":LET Y$="MAX"
30 LET X$="TIEMPO":LET M$=" ":GOSUB 2000
40 LET X=0:LET N=0
50 GOSUB 140
60 LET X=X+DX:LET Y=LL
70 GOSUB 2200
80 FOR T=1 TO FT:NEXT T
90 IF X<1000 THEN GOTO 50
100 LET M$="APRIETA UNA TECLA":GOSUB 2400
* 110 GET I$
120 IF I$=" " THEN GOTO 110
130 GOTO 20
* 140 LET LL=1000-4*PEEK (54279)
150 RETURN
```

La computadora visualiza un gráfico de las variaciones de la luz, pero no ofrece ninguna lectura luminosa.



1 Cosas que hacer




Apunta el sensor a fuentes luminosas diferentes y usa el gráfico para comparar su intensidad.

2




Si apuntas el sensor hacia una ventana el gráfico mostrará cuándo pasa una persona ya que ésta bloquea la luz durante un instante.

3



Usa el sensor para grabar los cambios luminosos de una habitación durante un día de verano y un día de invierno. Tienes que hacer que la computadora trace el gráfico muy lentamente incrementando el valor de FT en la línea 10 del programa.

4



Mueve el sensor por una pantalla de TV y observa los cambios que obtienes en el gráfico.


Almacenamiento de información

En esta parte del libro encontrarás programas para hacer que tu computadora almacene y analice información que obtengas en experimentos o encuestas. El programa que se describe en las páginas siguientes es una base de datos. Permite que almacenes resultados de una encuesta o de un experimento.

También puedes usarlo para guardar un archivo de una colección o de un hobby, como puede ser el de datos de aviones. El listado del programa está en las páginas 35-36. Debajo tienes algunos detalles sobre el funcionamiento de una base de datos.

Sobre la base de datos

El programa base de datos almacena información en archivos (files). Un archivo es como una página de una agenda. Está dividido en cuatro «columnas» en las que se almacena diferente información.



Das a cada columna un título. Luego introduces los datos dando a la computadora información para colocar en cada columna.

Una vez que la computadora tiene un

archivo con varios datos puede buscar en su interior alguna información en particular, como puede ser las piedras que contengan el mineral calcita o los aviones que tengan cuatro motores jet.

Uso de base de datos

Estas son algunas recomendaciones para ejecutar el programa base de datos. Cuando hayas introducido un archivo grábalo en cinta para no perderlo al apagar la computadora o al volver a ejecutar el programa. Puedes hacer tantos archivos como quieras pero sólo podrás cargar y usar uno cada vez.

1 El menú

MENU PRINCIPAL

1. CARGAR ARCHIVO
2. CREAR NUEVO ARCHIVO
3. INTRODUCIR DATOS
4. VER DATOS
5. CAMBIAR DATOS
6. GRABAR ARCHIVO
7. BUSCAR ARCHIVO

ELIGE UNA OPCION

?

Cuando ejecutas el programa la computadora visualiza un «menú» que contiene todas las funciones que puede llevar a cabo. Para escoger una función escribes su número y aprietas RETURN. Cuando estés usando el programa puedes volver al menú apretando RETURN.

3 Crear un nuevo archivo

INTRODUCE LOS TITULOS

TITULO 1

?NOMBRE

TITULO 2

?COLOR

La opción 2 es para crear un nuevo archivo. Empiezas por dar a la computadora los títulos de las cuatro columnas. Cada uno de ellos no debe sobrepasar las diez letras o espacios, por lo que quizás tengas que abreviarlos.

5 Ver los datos

VER DATOS

AHORA EN SERIE DE DATOS 2

NOMBRE	CRETA
COLOR	BLANCO
FECHA DE LOCALIZACION	5/7/82
MINERAL	CALCITA

La opción 4 te permite observar una a una todas las series de datos del archivo. Para avanzar a la siguiente serie de datos aprieta la barra espaciadora.

2 Cargar un archivo

CARGAR ARCHIVO

NOMBRE DEL ARCHIVO? PIEDRAS

APRIETA PLAY EN EL CASSETTE

Tu computadora visualizará un mensaje similar a éste.



Si quieres ver un archivo que ya tienes creado y almacenado en cinta elige la opción 1 para cargarla en la computadora. Debes darle a la computadora el nombre del archivo y después apretar PLAY en el cassette.

4 Introducir datos

INTRODUCE DATO

AHORA EN SERIE DE DATOS 2


NOMBRE

?ARENISCA

COLOR

?ROJO/AMARILLO

Puedes introducir varias palabras pero no las separes con comas o punto y coma.



Para introducir datos en un archivo elige la opción 3.

La computadora visualiza los títulos de las columnas una a una y pide información para introducir en ellas.

6 Cambiar datos

CAMBIAR DATOS

AHORA EN SERIE DE DATOS 2

NOMBRE

?CRETA

COLOR

?BLANCO/AMARILLO


Puedes cambiar los datos que ves apretando RETURN para volver al menú y seleccionando la opción 5. Esto te permite volver a introducir una serie de datos completa.

7 Grabar el archivo

GRABAR ARCHIVO

NOMBRE DEL ARCHIVO?
PIEDRAS
APRIETA RECORD Y PLAY
EN EL CASSETTE

Este mensaje puede aparecer de forma diferente en tu computadora o ni siquiera aparecer (aunque tienes que hacerlo).



Asegúrate de que la cinta está en el punto donde quieres empezar a grabar.

Es aconsejable grabar cada archivo nada más completarlo. Si no, puede que te olvides y lo pierdas al volver a ejecutar el programa o al apagar la computadora.

Para grabar un archivo introduces un nombre y aprietas RECORD y PLAY en el cassette. Escribe el nombre de los archivos en una etiqueta no vaya a ser que te olvides.

8 Buscar

1. La opción 7 hace que la computadora busque información en los archivos. A la derecha puedes ver cómo decir a la computadora qué información buscar.

2. También debes indicarle a la computadora qué visualizar cuando encuentre la información. Por ejemplo, si está buscando motores de tres jet, debes decirle si quieres que visualice el tipo de avión, la compañía para la que vuelan o su velocidad máxima —o todo junto.

3. Cuando la computadora ha buscado en los archivos visualiza la información por series de datos. Debes apretar la barra espaciadora para pasar de una serie de datos a otra.

BUSCAR ARCHIVO

- 1 NOMBRE
- 2 MOTORES
- 3 COMPANIA AEREA
- 4 VELOCIDAD MAXIMA

DATO BUSCADO

- ??
MOTORES? DOS JET
INFO REQUERIDA
?1
?3
?

Aprieta RETURN para saltarte una columna.

En primer lugar introduce el número de columna en el que se halla la información. Luego lo que busques en la computadora.



BUSCANDO:


MOTOR DOS JET

SERIE DE DATOS 3 DC 9


APRIETA BARRA ESPACIADORA

BUSQUEDA FINALIZADA


1 SERIE DE DATOS LOCALIZADOS




Bases de datos para hacer



Si te interesan las plantas o las setas podrías crear una base de datos con sus nombres comunes, la especie a la que pertenecen, dónde y cuándo los viste.



Una base de datos alimenticia puede servirte para seguir una dieta equilibrada. Puedes grabar las vitaminas, proteínas y grasas que tiene cada alimento.



Crea una base de datos con los años de los coches, las marcas, el número de ocupantes y el sexo del conductor. Encontrarás conclusiones como que las mujeres conducen coches más nuevos que los hombres.

Programa base de datos

Este es el programa para la base de datos. Las líneas son largas y complicadas por lo que debes copiarlas con mucho cuidado y


corregirlas antes de apretar RETURN. No te olvides de hacer las conversiones necesarias para tu computadora (ver páginas 42-45).

```

10 GOSUB 830
20 LET T$="MENU PRINCIPAL".GOSUB 710
30 PRINT:PRINT "1. CARGAR ARCHIVO":
  PRINT "2. CREAR NUEVO ARCHIVO"
40 PRINT "3. INTRODUCIR DATOS":PRINT "4. VER DATOS"
50 PRINT "5. CAMBIAR DATOS":PRINT "6. GRABAR ARCHIVO"
60 PRINT "7. BUSCAR ARCHIVO":PRINT:PRINT "ELIGE UNA OPCION"
70 GOSUB 750:LET CH=KP-48
80 IF CH<1 OR CH>ML THEN GOTO 20
* 90 ON CH GOSUB 500,780,110,460,160,500,180
100 GOTO 20
110 IF ULTIMO=NR-2 THEN RETURN
120 LET ULTIMO=ULTIMO+1
130 LET T$="INTRODUCE DATO".GOSUB 710
140 LET TP= ULTIMO.GOSUB 740
150 LET IP=ULTIMO.GOSUB 680: RETURN
160 LET T$="CAMBIAR DE DATOS".GOSUB 710.GOSUB 740
170 LET IP=TP.GOSUB 680:RETURN
180 LET T$="BUSCAR DATOS".GOSUB 710
190 FOR I=1 TO LP:LET P(I)=0:LET R$((NR-1)*LP+I)=" "
200 PRINT TAB(1);I;TAB(6);R$(I):NEXT I
210 PRINT:PRINT "DATO BUSCADO:"
220 LET A$=" ":INPUT A$:IF A$<"1" OR A$>"4" THEN GOTO 250
* 230 PRINT R$ (VAL(A$));":":INPUT W$
240 LET R$((NR-1)*LP+VAL(A$))=W$+":":IF W$>" " THEN GOTO 220
250 PRINT "INFO REQUERIDA:"
260 LET A$=" ":INPUT A$:IF A$<"1" OR A$>"4" THEN GOTO 280
270 LET P(VAL(A$))=1:GOTO 260
280 LET T$="BUSCANDO:".GOSUB 710
290 FOR I=1 TO 4
* 300 IF R$((NR-1)*LP+1)>" " THEN PRINT TAB (0);R$(I);TAB(12);R$((NR-1)*LP+I)
310 NEX I:PRINT:PRINT
320 LET C=0:FOR I=1 TO ULTIMO:LET FL=0
330 FOR J=1 TO LP
340 LET S$=R$(I*LP+J)
350 LET F$=R$((NR-1)*LP+J)
360 IF F$=" " THEN GOTO 420
370 LET F=0:LET LF=LEN(F$):LET LS=(LEN(S$)-LF)+1
380 FOR K=1 TO LS

```

El número de series de datos que puedes introducir en un archivo depende de la memoria de tu computadora. Cuando esté llena la computadora no admitirá más datos.



El listado continúa en la página siguiente.


```

* 390 IF MID$(S$,K,LF)=F$ THEN LET F=1:LET K=LS
400 NEXT K
410 IF F=0 THEN LET FL=1:LET J=LP
420 NEXT J:IF FL=0 THEN GOSUB 620
430 NEXT I
440 PRINT:PRINT "BUSQUEDA FINALIZADA":PRINT "LOCALIZADO ";C;" VECES"
450 GOSUB 750:RETURN
460 LET TP=1:LET T$="VER DATOS"
470 GOSUB 710:GOSUB 740:GOSUB 650:GOSUB 750
480 IF KP=32 AND TP<ULTIMO THEN LET TP=TP+1:GOTO 470
490 RETURN
500 LET T$="GRABAR ARCHIVO":IF CH=1 THEN LET T$="CARGAR ARCHIVO"
510 GOSUB 710:PRINT "NOMBRE DEL ARCHIVO:":INPUT F$
520 IF CH=6 THEN LET R$((ULTIMO+1)*LP+1)="**"
* 530 IF CH=6 THEN OPEN 1,1,1,F$
* 540 IF CH=1 THEN OPEN 1,1,0,F$
550 LET I=1
* 560 IF CH=6 THEN PRINT#1,R$(I)
* 570 IF CH=1 THEN INPUT#1,R$(I)
580 IF R$(I)<>"**" THEN LET I=I+1:GOTO 560
* 590 CLOSE 1
600 IF CH=1 THEN LET ULTIMO=INT(I/LP)-1
610 RETURN
620 PRINT "SERIE DE DATOS ";I:LET C=C+1:FOR K=1 TO LP
630 IF P(K)=1 THEN PRINT TAB(12);R$(I*LP+K)
640 NEXT K:PRINT:PRINT "APRIETA BARRA ESPACIADORA":PRINT:GOSUB 750:RETURN
650 FOR K=1 TO LP
* 660 PRINT TAB(1);R$(K);TAB(15);R$(TP*LP+K)
670 NEXT K:RETURN
680 FOR I=1 TO LP:PRINT R$(I)
690 LET A$="":INPUT A$:IF A$=" " THEN LET A$="-"
700 LET R$(I*LP+I)=A$+" ":NEXT I:RETURN
* 710 CLS:PRINT:PRINT
720 PRINT "BASE DE DATOS":PRINT "_____":PRINT
730 PRINT T$:PRINT:RETURN
740 PRINT:PRINT "AHORA EN SERIE DE DATOS ";TP:PRINT:RETURN
* 750 GET I$
760 IF I$=" " THEN GOTO 750
* 770 LET KP=ASC(I$):RETURN
780 LET T$="CREAR NUEVO ARCHIVO":GOSUB 710
790 PRINT "INTRODUCE LOS TITULOS":PRINT:FOR I=1 TO LP
800 PRINT "TITULO ";I:INPUT A$
810 IF LEN(A$)>10 THEN PRINT "DEMASIADO LARGO":GOTO 800
820 LET R$(I)=A$:NEXT I:GOSUB 850:RETURN
830 LET NR=30:LET LP=4:LET ML=7
* 840 DIM R$(NR*LP),P(LP)
850 LET ULTIMO=0:LET TP=1:RETURN

```


Observación de los resultados

Los experimentos y las encuestas suelen producir grandes cantidades de cifras. Para poder entender qué significan todas estas cifras, los científicos las resumen y

las generalizan de varias formas. Este programa y el de la página siguiente sirven para hacer esto.

Programa para hallar el valor promedio


```
* 10 CLS
20 LET LT=100
30 DIM X(LT+1)
40 LET N=1
50 PRINT "VALOR ";N;" ":INPUT A$
60 IF A$="E" THEN GOTO 100
70 LET X(N)=VAL(A$):LET N=N+1
80 IF N<LT+1 THEN GOTO 50
90 PRINT "NO HAY MAS ESPACIO"
100 PRINT "FIN DE LOS DATOS"
110 LET N=N-1
120 GOSUB 150:GOSUB 210
130 GOSUB 250:GOSUB 290
140 STOP
150 FOR I=1 TO N-1:LET LW=X(I):LET P=I
160 FOR J=I+1 TO N
170 IF X(J)<LW THEN LET LW=X(J):LET P=J
180 NEXT J
190 LET T=X(I):LET X(I)=X(P):LET X(P)=T
200 NEXT I:LET X(N+1)=X(N)+1:RETURN
210 LET T=0:FOR I=1 TO N
220 LET T=T+X(I):NEXT I
230 PRINT "MEDIA = ";T/N
240 RETURN
250 LET MX=N/2:LET MD=X(INT(MX+1))
260 IF MX=INT(MX) THEN LET MD=(X(MX)+X(MX+1))/2
270 PRINT "MEDIANA = ";MD
280 RETURN
290 LET SP=1:LET LP=1:LET MD=0
300 LET P=-1
310 LET P=P+1
320 IF X(SP)=X(SP+P) THEN GOTO 310
330 IF P>LP THEN LET MD=X(SP):LET LP=P
340 LET SP=SP+P
350 IF SP<N+1 THEN GOTO 300
360 IF MD>0 THEN PRINT "MODA = ";MD
370 RETURN
```

Un valor promedio es una forma de generalizar cifras. Por ejemplo, puedes oír hablar del promedio de litros de lluvia caídos durante un día, o del promedio de beneficios de una empresa.


En realidad existen tres tipos de promedios, —la media, la mediana y la moda— que pueden ofrecer tres valores diferentes trabajando con un mismo grupo de cifras.

A los investigadores suele interesarles el valor promedio ya que éste es el más significativo de toda la información con la que trabajan.

Por ejemplo, si quieres hallar el número promedio de miembros de una familia te interesará más la moda que la media, ya que ésta puede ofrecerte un número como 3,5 personas.



Para usar el programa introduce varias cifras una a una, apretando RETURN después de cada una. Finalmente introduce una E para indicar que has terminado.



Este programa calcula estos tres promedios, por lo que podrás compararlos. La media se halla sumando todas las cifras y dividiendo el resultado entre el número de cifras. La mediana es el valor obtenido extrayendo la cifra media de la lista de cifras colocadas por orden de tamaño. La moda es la cifra que se repite más veces.

Programa correlación

Una correlación consiste en descubrir si dos series de mediciones están relacionadas entre sí. Por ejemplo, ¿está relacionada la cifra de crecimiento de una planta con la cantidad de luz que recibe? ¿Crecen las orejas de las personas al envejecer?


El programa inferior hace que tu computadora compruebe mediciones para ver si están relacionadas. En la página opuesta tienes algunas ideas para tomar mediciones y correlaciones.




Introduce las mediciones por pares, por ejemplo, la edad de una persona y el tamaño de sus orejas.



El programa te dice si los grupos de mediciones están correlacionados positiva o inversamente. En la página opuesta encontrarás qué significa esto.



El programa también te indica la fiabilidad del resultado. Esto depende de cuántos pares de mediciones introduces.




```
* 10 CLS
20 DIM A(100,2)
30 PRINT "PROGRAMA CORRELACION"
40 PRINT
50 PRINT "INTRODUCE LOS NUMEROS POR PARES"
60 PRINT "INTRODUCE UNA E"
70 PRINT "CUANDO HAYAS TERMINADO"
80 LET S=0:LET F=0:LET FS=0:LET FF=0
90 LET SS=0:LET N=0
100 LET N=N+1
110 PRINT
120 INPUT "PRIMERO :";A$
130 IF A$="E" THEN GOTO 190
140 LET A(N,1)=VAL(A$)
150 INPUT "SEGUNDO :";A(N,2)
160 LET F=F+A(N,1)
170 LET S=S+A(N,2)
180 GOTO 100
190 LET N=N-1:LET MF=F/N:LET MS=S/N
200 FOR Y=1 TO N
210 LET FS=FS+((A(Y,1)-MF)*(A(Y,2)-MS))
220 LET FF=FF+((A(Y,1)-MF)*(A(Y,1)-MF))
230 LET SS=SS+((A(Y,2)-MS)*(A(Y,2)-MS))
240 NEXT Y
250 LET R=FS/SQR(SS*FF)
260 LET D$="POSITIVAMENTE"
270 IF R<0 THEN LET D$="INVERSAMENTE"
280 LET C$="RAZONABLEMENTE"
290 IF ABS(R)>0.7 THEN LET C$="FUERTEMENTE"
300 IF ABS(R)<0.3 THEN LET C$="POBREMENTE"
310 LET E$="RELATIVA"
320 IF N<10 THEN LET E$="POCA"
330 IF N>60 THEN LET E$="MUCHA"
340 PRINT "LOS VALORES ESTAN ";C$:PRINT D$;" CORRELACIONADOS."
350 PRINT "YA QUE HAY ";N;" PARES"
360 PRINT " PUEDES TENER ";E$
370 PRINT "CONFIANZA EN EL RESULTADO"
380 STOP
```


Más sobre correlaciones

1


Si tus orejas crecen según vas envejeciendo quiere decir que el tamaño de las orejas y la edad están positivamente correlacionados.



Las mediciones pueden estar o positiva o inversamente correlacionadas. Si lo están positivamente quiere decir que al aumentar un grupo de cifras también lo hace el otro.

2

Si tus orejas se hacen más pequeñas según envejeces, el tamaño de las orejas y la edad están inversamente relacionadas.



Las mediciones están inversamente relacionadas cuando un grupo de cifras aumenta al disminuir el otro. Haz una investigación sobre la relación edad, tamaño de las orejas y estudia si están correlacionadas.


Correlaciones para experimentar

Aquí tienes algunas ideas sobre cosas que puedes medir para averiguar si están correlacionadas. Si dos cosas están correlacionadas suele querer decir que una es causada por la otra, pero no necesariamente. Ambas pueden ser

causadas por un tercer factor, o la correlación puede ser una coincidencia. Puedes, por ejemplo, encontrar una correlación entre la altura de varias personas y sus resultados en un examen de francés.

1

Haz una señal en el hilo mientras la cometa esté en el aire. Luego mídela con la cinta métrica.




Investiga a qué altura vuelan diferentes cometas (puedes hacerte una idea aproximada midiendo el hilo que sueltas). Luego pégalas para ver si el peso de una cometa está relacionado con lo alto que vuela.

2



¿Está relacionada la altura de una persona con lo que puede beber? Mide a tus amigos y haz un concurso de beber.

3



En los periódicos puedes encontrar grandes cantidades de información que puedes probar a correlacionar. Por ejemplo, puedes estudiar la relación entre el tipo de cambio y el tipo de interés, o la temperatura diaria.

¿Probable o improbable?

Antes de que un investigador acepte el resultado de un experimento o de una encuesta como prueba de una teoría, tiene que averiguar qué posibilidad existe de que puedan obtenerse dichos resultados por suerte. El programa inferior es un ejemplo de una prueba estadística que comprueba la posibilidad de que se dé un determinado resultado.

Por ejemplo, imagínate que estás comprobando si un dado que has comprado está bien equilibrado (es decir, que hay las

mismas posibilidades de que caiga sobre cualquiera de sus seis caras).

Normalmente al tirar un dado la posibilidad de sacar un seis es de una entre seis. Si pruebas a tirar el dado 12 veces debes esperar que salgan dos seises. Digamos que sacas cinco seises, ¿demuestra esto que el dado está descompensado?, ¿o es más lógico pensar que el resultado se haya dado por casualidad? Ejecuta el programa para averiguarlo.

Programa de casualidades

```
* 10 CLS
20 PRINT:PRINT
30 PRINT "CUANTOS INTENTOS":INPUT N
40 PRINT "CUANTAS VECES"
50 PRINT "ESPERAS QUE SALGA"
60 PRINT "EL RESULTADO":INPUT E
70 PRINT "CUANTAS VECES"
80 PRINT "SALIO EL RESULTADO":INPUT R
90 IF E>=N OR E<=0 OR R>N OR R<0 THEN PRINT "REVISA!":GOTO 30
100 IF R<E THEN LET TL=R+1:LET BL=-1
110 IF R>=E THEN LET TL=N+1:LET BL=R-1
120 LET P=E/N
130 LET FR=P/(1-P)
140 LET PP=(1-P) ^ N
150 LET MP=0:IF BL=-1 THEN LET MP=PP
160 FOR I=1 TO N
170 LET PN=((N-I+1)/I)*PP*FR
180 LET PP=PN
190 IF I<TL AND I>BL THEN LET MP=MP+PN
200 NEXT I
210 PRINT:PRINT
220 LET M$="MAS":IF BL=-1 THEN LET M$="MENOS"
230 PRINT "LA POSIBILIDAD DE UN RESULTADO"
240 PRINT "DE ";R;" (O ";M$;" ) ES"
250 IF MP=0 THEN PRINT "DEMASIADO PEQUEÑA":GOTO 280
260 LET CH=1/MP:GOSUB 290
270 PRINT "1 IN ";CH;M$
280 PRINT:STOP
290 LET M$=" "
300 IF CH>1E6 THEN LET CH=CH/1E6:LET M$=" MILLON"+M$:GOTO 300
310 IF CH>1E3 THEN LET CH=CH/1E3:LET M$=" MIL"+M$:GOTO 310
320 LET CH=INT(10*CH+0.5)/10
330 RETURN
```


Ejecutar el programa

1

CUANTAS VECES HICISTE
EL EXPERIMENTO
?12

Primero introduce el número de veces que tiraste el dado.

2

CUANTAS VECES
ESPERAS QUE SALGA
EL RESULTADO
?2
CUANTAS VECES
SALIO EL RESULTADO
?5


A continuación dile a la computadora cuántas veces esperabas que saliese el resultado (un seis) y cuántas veces salió realmente.

3


LA POSIBILIDAD DE UN RESULTADO
DE 5 (O MAS) ES
1 IN 27.5

Usando la información que has introducido, la computadora calcula que la posibilidad de sacar cinco (o más) seises es de 1 en 27. Por tanto el resultado es muy improbable y es posible que el dado esté descompensado.

Los científicos normalmente buscan posibilidades de obtener un resultado menor que 1 en 20 (es decir, 1 en 30 y 1 en 40) antes de aceptar que el resultado no salía por casualidad.



Experimento con cartas



Puedes usar el programa para comprobar los resultados de cualquier experimento del que conoces la probabilidad del fenómeno que estás estudiando. Por ejemplo, si un amigo elige una carta de la baraja, tus posibilidades de acertar de qué palo es, son de uno en

cuatro. Si lo realizas 20 veces debes esperar obtener cinco aciertos. ¿Cuáles son tus posibilidades de acertar nueve veces el palo, o doce veces? Usa el programa para averiguarlo.

Conversiones para los programas

En esta página y en las siguientes tienes unas líneas de conversión que te permitirán adaptar los programas a las distintas computadoras. Para buscar una línea de conversión busca en la sección

BBC

Página 11

Programa de pulsaciones

```
390 LET I$=INKEY$(0)
500 LET FT=22000:LET PT=2000
```

Página 15

Programa simulador de trayectos

En todo el programa debes sustituir rnd por RND(1) tal y como se muestra aquí.

```
650 LET BJ=SB+INT
(DB*(RND(1)+RND(1))*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT (DB*
(RND(1)+RND(1))*0.5)
```

Cambios para esperar

```
50 LET WT=INT(RND(1)*FB)
```

Página 21

Dirigir una línea aérea

```
820 LET PS=POS-28
```

Página 26

Programa para la resistencia térmica

```
1000 LET V=1.8*ADVAL(1)/65520
1010 LET R=V/(5-V)
1020 LET T=1/((LN(R/K2))
/K1+1/273)-273
```

Página 28

Líneas de gráficos

```
30 LET FT=100:LET DX=5
140 LET I$=INKEY$(0)
```

Página 31

Programa para el sensor luminoso

```
10 LET FT=100:LET DX=5
110 I$=INKEY$(0)
140 LET LL=1000-INT(ADVAL(1)/66)
```

Página 35

Programa base de datos

```
530 IF CH=6 THEN X=OPENOUT F$
540 IF CH=1 THEN X=OPENIN F$
560 IF CH=6 THEN PRINT # X,R$(I)
570 IF CH=1 THEN INPUT# X,R$(I)
590 CLOSE # X
750 LET I$=INKEY$(0)
```

para tu computadora el número de página y el nombre del programa. Si no hay conversión puedes copiar la línea tal y como viene en el programa.

Electron

Página 11

Programa de pulsaciones

```
390 LET I$=INKEY$(0)
500 LET FT=7750:LET PT=1000
```

Página 15

Programa simulador de trayectos

En todo el programa sustituye rnd por RND(1) tal y como se muestra aquí.

```
650 LET BJ=SB+INT
(DB*(RND(1)+RND(1))*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT (DB*
(RND(1)+RND(1))*0.5)
```

Cambios para esperar

```
50 LET WT=INT(RND(1)*FB)
```

Página 21

Dirigir una línea aérea

```
820 LET PS=POS-28
```

Página 35

Programa base de datos

```
530 IF CH=6 THEN X=OPENOUT F$
540 IF CH=1 THEN X=OPENIN F$
560 IF CH=6 THEN PRINT # X,R$(I)
570 IF CH=1 THEN INPUT# X,R$(I)
590 CLOSE # X
750 LET I$=INKEY$(0)
```

VIC 20

Página 7

Programa para jugar a cara o cruz

```
10 PRINT CHR$(147)
```

Página 8

Pelota que rebota

```
10 PRINT CHR$(147):PRINT
```

Página 11

Programa de pulsaciones

```
420 PRINT CHR$(147)
440 POKE 36878,15:POKE 36876,200
442 FOR W=1 TO 100:NEXT W
444 POKE 36876,0
480 LET C(1)=2:LET C(2)=7:
LET C(3)=5
500 LET FT=9000:LET PT=1000
```


VIC 20, continuación

Página 15

Programa simulador de trayectos

```
10 PRINT CHR$(147)
```

En todo el programa sustituye
rnd por RND(1) tal y como se
muestra aquí.

```
650 LET BJ=SB+INT  
      (DB*(RND(1)+RND(1))*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT (DB*  
  (RND(1)+RND(1))*0.5)  
90 PRINT CHR$(147)  
140 FOR J=1 TO F(N)/2:PRINT  
    " ";:NEXT J
```

Cambios para esperar

```
50 LET WT=INT(RND(1)*FB)
```

Página 21

Dirigir una línea aérea

```
820 LET PS=POS(0)-13  
880 PRINT CHR$(147)
```

Página 26

Programa para la resistencia térmica

```
1000 LET TR=1.84*PEEK(36872)
```

Página 28

Líneas de gráficos

```
30 LET FT=10:LET DX=20
```

Página 31

Programa para el sensor luminoso

```
10 LET FT=10:LET DX=20  
140 LET LL=1000-4*PEEK(36872)
```

Página 35

Programa base de datos

```
710 PRINT CHR$(147):PRINT:PRINT
```

Página 37

Programa para hallar el valor promedio

```
10 PRINT CHR$(147)
```

Página 38

Programa correlación

```
10 PRINT CHR$(147)
```

Página 40

¿Probable o improbable?

```
10 PRINT CHR$(147)
```

Commodore 64

Página 7

Programa para jugar a cara o cruz

```
10 PRINT CHR$(147)
```

Página 8

Pelota que bota

```
10 PRINT CHR$(147):PRINT
```

Página 11

Programa de pulsaciones

```
420 PRINT CHR$(147)  
440 POKE 54296,15:POKE 54277,0  
442 POKE 54278,240:POKE 54276,33  
444 POKE 54272,75:POKE 54273,39  
446 FOR W=1 TO 100:NEXT W  
448 POKE 54276,32:POKE 54296,0  
480 LET C(1)=2:LET C(2)=7:LET  
    C(3)=5
```

Página 15

Programa simulador de trayectos

```
10 PRINT CHR$(147)
```

En todo el programa sustituye rnd
por RND(1) tal y como se muestra.

```
650 LET BJ=SB+INT (DB*  
  (RND(1)+RND(1))*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT (DB*  
  (RND(1)+RND(1))*0.5)  
90 PRINT CHR$(147)
```

Cambios para esperar

```
50 LET WT=INT(RND(1)*FB)
```

Página 21

Dirigir una línea aérea

```
880 PRINT CHR$(147)
```

Página 35

Programa base de datos

```
710 PRINT CHR$(147):PRINT:PRINT
```

Página 37

Programa para hallar el valor promedio

```
10 PRINT CHR$(147)
```

Página 38


Programa correlación

```
10 PRINT CHR$(147)
```

Página 40

¿Probable o improbable?

```
10 PRINT CHR$(147)
```



Apple

Página 7

Programa para jugar a cara o cruz
10 HOME

Página 8

Pelota que bota
10 HOME:PRINT

Página 11

Programa de pulsaciones

```
390 LET I$="":IF PEEK(-16384)>127
    THEN GET I$
420 HOME
480 LET C(1)=43:LET C(2)=46:LET C(3)=42
500 LET FT=9700:LET PT=1000
```

En todo el programa cambia rnd por RND(1). Ej.:

Página 15

Programa simulador de trayectos

```
10 HOME
650 LET BJ=SB+INT (DB*(RND(1)
    +RND(1))*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT
    (DB*(RND(1)+RND(1))*0.5)
90 HOME
120 PRINT TAB(1);B+N;TAB(5);" ";
```

Cambios para esperar

```
50 LET WT=INT(RND(1)*FB)
120 PRINT TAB(1);B+N;"-";
    B+N+1;" ";
```

Página 21

Dirigir una línea aérea

```
880 HOME
```

Página 35

Programa base de datos

```
530 IF CH=6 THEN PRINT CHR$(4);
    "OPEN"+F$:PRINT CHR$(4);
    "WRITE"+F$
540 IF CH=1 THEN PRINT CHR$(4);
    "OPEN"+F$:PRINT CHR$(4);
    "READ"+F$
560 IF CH=6 THEN PRINT R$(I)
570 IF CH=1 THEN INPUT R$(I)
590 PRINT CHR$(4);"CLOSE"+F$
710 HOME:PRINT:PRINT
750 I$="":IF PEEK(-16384)>127
    THEN GET I$
```

Página 37

Programa para hallar el valor promedio

```
10 HOME
```

Página 38

Programa correlación

```
10 HOME
```

Página 40

¿Probable o improbable?

```
10 HOME
```

TRS-80

Página 7

Programa para jugar a cara o cruz
70 LET X=RND(0)

Página 11

Programa de pulsaciones

```
390 LET I$=INKEY$
440 SOUND 100,1
480 LET C(1)=4:LET C(2)=2:LET
    C(3)=1
500 LET FT=8700:LET PT=1000
```

Página 15

Programa simulador de trayectos

En todo el programa sustituye rnd por RND(0) Ej.

```
650 LET BJ=SB+INT
    (DB*(RND(0)+RND(0))*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT (DB*(RND(0)
    +RND(0))*0.5)
135 IF N=15 THEN IF INKEY$=
    "" THEN GOTO 135
```

Añade esta línea.

Cambios para esperar

```
50 LET WT=INT(RND(0)*FB)
```

Página 21

Dirigir una línea aérea

```
820 LET PS=POS(0)-24
```

Página 26

Programa para la resistencia térmica

```
1000 LET V=JOYSTK(0)*0.0715+0.25:
    LET R=V/(5-V)
1010 LET T=1/((LOG(R/K2))/K1+1
    /273)-273
1020 RETURN
```

Página 28

Líneas de gráficos

```
30 LET FT=30:LET DX=10
140 LET I$=INKEY$
```

No hay línea 1030.

Página 31

Programa para el sensor luminoso

```
10 LET FT=30:LET DX=10
110 LET I$=INKEY$
140 LET LL=1000-16*JOYSTK(0)
```

Página 35

Programa base de datos

```
530 IF CH=6 THEN OPEN "O", # -1,F$
540 IF CH=1 THEN OPEN "I", # -1,F$
560 IF CH=6 THEN PRINT # -1,R$(I)
570 IF CH=1 THEN INPUT # -1,R$(I)
590 CLOSE # -1
750 LET I$=INKEY$
```

Letra O



Spectrum

Página 7

Programa para jugar a cara o cruz

```
70 LET X=RND
```

Página 11

Programa de pulsaciones

```
390 LET I$=INKEY$
440 BEEP 0.1,5
480 LET C(1)=2:LET C(2)=6:LET C(3)=4
500 LET FT=1950:LET PT=300
```

Página 15

Programa simulador de trayectos

```
650 LET BJ=SB+INT(DB*(RND+RND)*0.5)
```

Página 19

Programa del viaje en autobús

```
50 LET BJ=SB+INT(DB*(RND+RND)*0.5)
```

Cambios para esperar

```
50 LET WT=INT(RND*FB)
```

Página 21

Dirigir una línea aérea

```
810 LET M$=P$(L)
820 LET PS=8-PEEK(23688)
```

Página 35

Programa base de datos

```
90 GOSUB 500*(CH=1)+780*(CH=2)+110*(CH=3)+460*(CH=4)+160*
  (CH=5)+500*(CH=6)+180*(CH=7)
225 PRINT A$ ← Añade esta línea.
230 PRINT R$(VAL(A$))( TO 10);“.”;:INPUT W$:PRINT W$
265 PRINT A$ ← Añade esta línea.
300 IF R$((NR-1)*LP+I)(1)<>“ ” THEN PRINT
  TAB(0);R$(I);TAB(12);R$((NR-1)*LP+I)
345 IF S$(LEN(S$)-1 TO LEN(S$))=“ ” THEN LET S$=S$( TO
  LEN(S$)-1):GOTO 345 ← Dos espacios aquí.
355 IF F$(LEN(F$))=“ ” AND LEN(F$)>1 THEN LET F$=F$( TO
  LEN(F$)-1):GOTO 355
357 IF F$=“ ” THEN LET F$=“ ”
365 LET F$=F$+“ ”
390 IF S$(K TO K+LF-1)=F$ THEN LET F=1:LET K=LS
515 PRINT F$ ← Añade esta línea.
530 IF CH=6 THEN SAVE F$ DATA R$(I)
540 IF CH=1 THEN LOAD F$ DATA R$(I)
```

Añade estas líneas.

No copies las líneas 550, 560, 570, 580 y 590.



```
595 LET I=1 ← Añade estas líneas.
597 IF R$(I)( TO 2)<>“*” THEN LET I=I+1:GOTO 597
660 PRINT TAB(1);R$(K)( TO 10);TAB(15);R$(TP*LP+K)
695 PRINT A$
750 LET I$=INKEY$
770 LET KP=CODE(I$):RETURN ← Añade estas líneas.
805 PRINT A$
840 DIM R$(NR*LP,30):DIM P(LP)
```

Página 37

Programa para hallar el valor promedio

```
55 PRINT A$
```

En todo el programa sustituye rnd por RND. Por ejemplo, la línea 650 debe quedar así.



Rutinas de gráficos

En estas dos páginas hay versiones de la rutina de gráficos que debes añadir al programa Pelota que rebota, al programa de pulsaciones y a los programas de la Resistencia Térmica y del Sensor Luminoso. Asegúrate de usar la versión

correspondiente a tu computadora. Las líneas son complicadas, con multitud de números y símbolos, por lo que debes copiarlos cuidadosamente y comprobar cada línea antes de apretar RETURN.

Apple

```
2000 HOME
2010 FOR I=1 TO 22:VTAB(I):PRINT " !":NEXT I
2020 FOR I=1 TO 39:VTAB(23):HTAB(I):PRINT "-":NEXT I
2030 VTAB(23):HTAB(1):PRINT "0"
2040 VTAB(1):PRINT Y$:TAB(10):T$
2050 VTAB(11):PRINT M$
2060 VTAB(24):HTAB(34):PRINT X$;
2070 LET DC=43:RETURN
2200 VTAB(1):HTAB(34):PRINT N;" "
2205 IF X<0 OR Y<0 OR X>1000 OR Y>1000 THEN RETURN
2210 LET QX=X/28:LET QY=Y/48
2220 VTAB(22-QY):HTAB(QX+3):PRINT CHR$(DC)
2230 RETURN
2400 VTAB(24):HTAB(5):PRINT "
2410 VTAB(24):HTAB(5):PRINT M$;
2420 RETURN
```

Deja 13 espacios aquí

↙ ";

BBC y Electron

```
2000 MODE 1:VDU 19,3,2,0;:COLOUR2:CLS:GCOL0,2:DC=1
2010 MOVE 0,100:DRAW 1200,100:MOVE 50,50:DRAW 50,1000
2020 PRINT TAB(10,1);T$:TAB(0,1);Y$:TAB(33,30);X$
2030 PRINT TAB(0,14);M$:TAB(0,28);"0":RETURN
2200 IF X<0 OR Y<0 THEN RETURN
2210 GCOL0,DC:PLOT 69,X+50,Y*0.88+100
2220 PRINT TAB(33,1);N;" "
2230 RETURN
2400 PRINT TAB(5,30);STRING$(13," ");
2410 PRINT TAB(5,30);M$;
2420 RETURN
```

Spectrum

```
2000 PAPER 0:CLS:INK 6:LET DC=2
2010 PLOT 0,15:DRAW 255,0
2020 PLOT 7,0:DRAW 0,175
2030 PRINT AT 0,0;Y$:AT 0,8;T$:AT 20,0;"0"
2040 PRINT AT 21,25;X$:AT 10,0;M$:RETURN
2200 INK 6:PRINT AT 0,27;N;" "
2210 IF X<0 OR Y<0 THEN RETURN
2220 LET QX=X/4+8:LET QY=Y/6.25+16
2230 IF QX>255 OR QY>175 THEN RETURN
2240 INK DC:PLOT QX,QY:INK 6:RETURN
2400 PRINT AT 21,5;"
2410 PRINT AT 21,5;M$:RETURN
```

Deja 13 espacios aquí

TRS-80

```

2000 CLS(0)
2010 FOR I=0 TO 29:SET(1,1,2):NEXT I
2020 FOR I=0 TO 63:SET(1,28,2):NEXT I
2030 PRINT@9,T$::PRINT@0,Y$;
2040 PRINT@507,X$::PRINT@448,"0":PRINT@192,M$;
2050 LET DC=4:RETURN
2200 IF X<0 OR Y<0 OR X>1000 OR Y>1000 THEN RETURN
2210 SET(X/16.5+2.27-Y/37.1,DC)
2220 PRINT@28,STR$(N);" ";
2230 RETURN
2400 PRINT@484,"
2410 PRINT@484,M$::RETURN

```

Deja 13 espacios aquí

Commodore 64

```

5 GOSUB 3000
2000 POKE 53281,0:POKE 53280,0
2010 PRINT CHR$(158);CHR$(147)
2020 PRINT HM$::FOR CT=1 TO SH-2:PRINT " C- M "
:NEXT
2030 PRINT "0 C- T "
2040 PRINT TAB(WS-5);X$;HM$;Y$;HM$;SPC(WS/5);T$
2050 PRINT HM$;LEFT$(CU$(SH-3)/2);M$
2060 DC=2:RETURN
2200 IF X<0 OR Y<0 OR X>1000 OR Y>1000 THEN RETURN
2210 QX=X/XK:QY=Y/YK
2220 ZX=INT(QX/2):ZY=INT(QY/2)
2230 BX=INT(QX-ZX*2):BY=INT(QY-ZY*2)
2240 QD=2↑(2-2*BY+BX):SN=BL+ZX-LC*ZY
2250 PN=PEEK(SN):FOR CT=0 TO 15
2260 IF QQ(CT)=PN THEN PN=QQ(CT OR QD):CT=15
2270 NEXT:POKE SN,PN:POKE SN+DM,DC
2280 PRINT HM$;SPC(WS-5);STR$(N);" ":RETURN
2400 PRINT HM$;LEFT$(CU$,SH-1);SPC(3);"
2410 PRINT HM$;LEFT$(CU$,SH-1);SPC(3);M$::RETURN
3000 DIM QQ(15):FOR I=0 TO 15:READ QQ(I):NEXT
3010 HM$=CHR$(19):CU$="":FOR I=1 TO 25:CU$=CU$+CHR$(17)
:NEXT
3020 LC=40:BL=1905:DM=54272
3030 XK=13:YK=22.2:SH=25:WS=40
3040 RETURN
3050 DATA 32,126,124,226,123,97,255,236,108,127,225,
251,98,252,254,160

```

Aquí debes apretar las teclas C y M al mismo tiempo.

Aprieta las teclas C y T al mismo tiempo 38 veces.

Deja 13 espacios aquí

VIC 20

Usa la versión del Commodore 64, pero cambia estas líneas.

```

2000 POKE 36879,8
2030 PRINT "0 C- T "
3020 LC=22:BL=8121:DM=30720
3030 XK=24:YK=24:SH=23:WS=22
















```

Aprieta las teclas C y T al mismo tiempo 20 veces

Indice

- Apple II, 3
 - conversiones para, 44
 - rutina de gráficos, 46
- Archivos en base de datos, 32-34
- Base de datos, 5, 32-34
 - de alimentos, 34
 - de aviones, 32
 - de coches, 34
 - de piedras, 32
 - de setas, 34
 - programa, 35-36
- BBC (B), 3, 24, 25, 26
 - conversiones para, 42
 - rutina de gráficos, 46
- Calibrar, 27
- Cara o cruz, programa, 7
- Chinche en un programa, 6
- Cinta aislante, 24, 25, 29, 30
- Clavijas para computadoras, 24, 25
- Commodore 64, 3, 24-25, 26
 - conversiones para, 43
 - rutina de gráficos, 47
- Conversiones para
 - programas, 6, 42-45
- Corregir programas, 6
- Correlaciones, 38-39
- Correlación inversa, 39
- Correlación positiva, 39
- Correlación programa, 38
- Cray-1, 4
- Din, clavija, 24, 25
- Dirección de una línea,
 - programa, 21-22
- Diseño de un cohete, 5
- Economía, 20
- Ejercicios para hacer, 10
- Electrodos, 4
- Electrón, 3
 - conversiones para, 42
 - rutina de gráficos, 46
- ENTER, 6
- Escape, tecla, 6
- Estañar, cómo, 29
- Experimentos
 - ácido-base, 28
 - con cartas, 41
 - con cometas, 39
 - con dados, 40-41
 - con la temperatura del cuerpo, 28
 - curva de enfriamiento, 28
 - de beber, 39
 - sobre la temperatura en una habitación, 28
- Exportaciones, 20
- Grabar
 - archivos en cinta, 34
 - programas, 3
- Gráficos, 10, 11, 24, 28, 31
- Gravedad, 9
- Investigación cerebral, 4
- Investigación criminal, 5
- Investigación operacional, 14
- Júpiter, gravedad en, 9
- Media, 37
- Mediana, 37
- Mercurio, gravedad en, 9
- Moda, 37
- Modelo, 3, 4, 5
 - cara o cruz, 7
 - económico, 20-23
 - pelota que bota, 8-9
 - trayecto, 14-19
- Modelos macro-económicos, 20
- Modelos micro-económicos, 20
- Neptuno, gravedad en, 9
- Porcentajes, 19
- Port analógico, 24, 26
 - clavijas, 24
- Port de control, 26
- Port para joystick, 26
- Posibilidades, programa, 40
- Programas, escribir y ejecutar, 6
- Pulsaciones, programa 11-12
- Pulso, cómo tomarlo, 13
- Resistencia, 24, 25
 - cómo calibrar, 27
- RETURN, 6
- RUN, 6
- Saturno, gravedad en, 9
- Sensores, 3, 4, 24, 31
 - de temperatura, 24-29
 - luminoso, 30-31
- Simulación de trayectos,
 - programas, 15-16
- Soldador, 24
- Soldar, 24, 25
 - cómo, 29
- Spectrum, 3
 - conversiones para, 45
 - rutina de gráficos, 46
- Tasa de inflación, 22, 23, 39
- Teoría de la gran explosión, 4
- Test sobre forma física, 13
- Tipo de cambio, 22, 23, 39
- Tipo de interés, 22, 23
- TRS-80 Colour Computer (32 K), 3, 19, 24, 25, 26
 - conversiones para, 44
 - rutina de gráficos, 47
- Urano, gravedad en, 9
- Valor promedio, 37
- Valor promedio, programa, 37
- Venus, gravedad en, 9
- VIC 20, 3, 24, 25, 26
 - conversiones para, 42-43
 - rutina de gráficos, 47

Colección Electrónica



distribuidor
exclusivo
cesma sa
C/ Aguacate, 25
28044 MADRID